
ПРЕДИСЛОВИЕ

Проектирование больших систем на уровне вентилей изжило себя.  Инженеры  движутся в направлении использования языков описания аппаратуры. Наиболее известными современными языками описания аппаратуры являются Verilog и VHDL. Эта книга  предназначена для разработчика,  ее авторами являются разработчики,  которые лучше знают, как учить проектированию на языке описания аппаратуры. Verilog создан в 1985 году в фирме Gateway.  Он используется более,  чем 10000 разработчиками таких фирм,  как Sun Microsystems, Apple Computer, Nexgen Microsystems, Motorola, Stardent.

Рей Вайс,

старший редактор  по  технологии,

газета Electronic Engineering Times

ВВЕДЕНИЕ

     Разработки становятся все сложнее,  поэтому  вентильный уровень описания  становится  все более непонятным,  вызывая необходимость описания проекта в более абстрактном виде. Как в 70-ые  годы  языки  высокого  уровня заменили ассемблерные языки, так в 90-ые годы языки  описания  аппаратуры заменят описание схем,  на  вентильном  уровне.  Средства логического синтеза будут выполнять  реализацию  на  вентильном  уровне. Включение языка описания аппаратуры и логического синтезатора в технику проектирования будет необходимостью.

     В настоящее время существуют два основных языка  описания аппаратуры  - Verilog и VHDL.  Оба они стали стандартами при проектировании цифровых схем.  Язык Verilog обеспечивает очень лаконичный и удобочитаемый синтаксис.

     Эта книга является результатом реального  опыта  работы на языке Verilog. Нашей целью является показать, как функционально описывать части аппаратуры,  используя подход проектирования сверху вниз. Все представленные в книге модели были промоделированы  и  верифицированы  с  помощью  программы моделирования Verilog-XL 1.5.

     Мы ожидаем от читателей знания логического проектирования. Полезен  опыт программирования на языках высокого уровня, таких как Си или других языках описания аппаратуры.  Модели в тексте упрощены в целях более ясного понимания.

     Хотя язык Verilog имеет конструкции для  проектирования на более низком уровне, а именно, на вентильном и переключательном уровнях, это не описано в этой книге.

Организация книги

     Книга начинается с введения в Verilog  и  заканчивается примером описания и моделирования большой подсистемы.

     В главе 2 основное внимание  уделено  конструкциям  для функционального моделирования языка Verilog. Приведен пример для иллюстрации метода проектирования сверху вниз.

     В главе 3 описывается компьютер с малым набором  команд (SISC  -  small  instruction set computer) и рассказывается, как моделировать процессор на СБИС на уровне  архитектуры  и на уровне регистровых передач.

     В главе 4 показывается, как моделировать различные блоки SISС и его центральный процессор.

     В главе 5 обсуждается кэш-память.

     В главе 6 рассматривается моделирование асинхронного интерфейса.

     В главе 7 представлен пример полного моделирования большого устройства.  Рассматривается подсистема контроллера флоппи-диска.

     В главе 8 даются полезные советы по моделированию и отладке.

Глава 1. ЗАЧЕМ НУЖНЫ ЯЗЫКИ ОПИСАНИЯ АППАРАТУРЫ

Тенденции развития методов проектирования

     Использование языков описания аппаратуры  (ЯОА)  в  логическом проектировании значительно расширилось  за  последние несколько лет.  Руководители  уже  не стоят перед дилеммой - проектировать на ЯОА или без них.

     Вместо этого их заботит выбор языка и его встраивание в систему проектирования. Разработчики теперь предпочитают выражать свой проект в функциональной или поведенческой форме, откладывая подробности реализации на более поздний этап проектирования.  Абстрактное представление помогает разработчикам рассматривать альтернативы в области архитектуры и обнаружить узкие места до начала детальной разработки.

     В результате  значительных успехов технологии постоянно увеличивается плотность кристалла  и  сложность  разработки. Достигнута плотность более миллиона транзисторов на кристалле и для того,  чтобы сделать такие сложные схемы доступными

человеку,  необходимо  выразить  функцию  на  языке высокого уровня, что скрывает подробности реализации. По этой же причине  языки  высокого  уровня  заменили ассемблерные языки в больших программных системах.

     Усиленное моделирование может обнаружить ошибки  проектирования до изготовления кристалла,  уменьшая таким образом количество итераций при проектировании.  Эффективные  ЯОА  и система моделирования  стали ценным инструментом для минимизации количества ошибок при проектировании и сделали возможным получение функционирующих кристаллов с первого раза.

     Тенденция к более объемным и более сложным проектам будет продолжаться. В 90-ые годы мы достигнем уровня миллиона вентилей. Разработчики обязательно  будут  проектировать  на ЯОА и оставят реализацию средствам логического синтеза.

Проектирование на ЯОА

     Использование ЯОА  может  обеспечить  несколько преимуществ.  Описание на ЯОА можно использовать  как  техническое задание  на  проект.  Преимущество использования формального языка,  такого как Verilog,  для описания заключается в том, что такое описание является полным и недвусмысленным. Описание на формальном языке является  "мягким"  по  сравнению  с "жестким"  описанием  схемы.  Представление на ЯОА позволяет легко обрабатывать текст на любом  текстовом  процессоре,  а базы данных схем обычно требуют графического редактора.

     Второй целью  использования ЯОА является моделирование. Моделирование разработки может  обнаружить  ошибки,  которые иначе обнаружились  бы только после изготовления аппаратуры. Моделирование может производится на нескольких  уровнях.  На функциональном уровне  система  описывается с использованием конструкций высокого уровня.  На логическом  уровне  система описывается иерархически,  когда  внизу  иерархии  находятся основные блоки.  Этот уровень может  включать  информацию  о временных задержках, давая возможность временного анализа.

     Третьей целью  использования  ЯОА  является  логический синтез. Существуют средства синтеза, которые могут по описанию разработки на ЯОА сгенерировать реализацию на вентильном уровне из библиотечных  элементов. Эти средства оптимизируют проект по задержке, объему схемы или по другой целевой функции. Существующие средства синтеза имеют некоторые ограничения, например,  они используют только некоторое подмножество этого языка и синтезированные схемы не могут быть такими  же эффективными, как  созданные  опытным разработчиком.  Тем не менее, синтез даже части схемы  может  сэкономить  время  и средства, предоставляя   возможность  разработки  упрощенной версии и   предварительной   оценки   быстродействия/площади кристалла.

     Наконец, ЯОА - это лучший способ документирования  проекта.  Хорошо прокомментированное описание на ЯОА может быть лучше и короче, чем схема на вентильном уровне.

Проектирование на языке Verilog

     Язык Verilog  прост  и элегантен.  Он имеет конструкции для описания элементов аппаратуры в краткой и  удобочитаемой форме. Аналогичное  описание  на языке VHDL может быть в два раза длиннее.

     Используя Verilog,  разработчик  должен  изучить только один язык для всех аспектов логического проектирования.  Моделирование требует  по меньшей мере функциональных моделей, иерархических структур,  тестовых векторов и  интерактивного взаимодействия человека и машины.  На Verilog это все достигается на одном языке.  Почти любой оператор,  который можно написать в программе, может быть выполнен с терминала.

     Verilog не только краткий и однородный,  его легко изучать. Он очень похож на язык Си. Так как Си - один из наиболее широко используемых языков  программирования,  то  большинству разработчиков  следовало  бы  знать его и им, было бы легче изучать Verilog.

Глава 2. ОСНОВЫ ЯЗЫКА ОПИСАНИЯ АППАРАТУРЫ  VERILOG

     В этой главе мы кратко знакомим с языком описания аппаратуры Verilog. Мы не собираемся дать полное описание языка. Полное описание Verilog есть в руководстве по языку.  В этой главе,  как и в книге в целом, уделено внимание функциональным  аспектам  языка.  В  приложении  А приводится синтаксис Verilog,  а в приложении В приводится список  ключевых  слов Verilog.

     В этой книге термин Verilog относится к языку  описания аппаратуры Verilog,   а   не   к   программе   моделирования Verilog-XL simulator.  Все ключевые  слова,  начинающиеся  с символа $,  являются командами Verilog-XL simulator и не являются частью языка Verilog.

     Verilog очень похож на язык программирования Си. Разработчики, знакомые с Си, легко овладеют языком Verilog. Как и Си, Verilog - язык со свободным форматом, учитывает регистр, все ключевые слова находятся на нижнем регистре. Для улучшения читаемости  можно использовать пробелы,  символы табуляции, начало с новой строки и комментарии.

ПОНЯТИЕ МОДУЛЯ

     Модуль (module) является основной единицей  в  Verilog. Он представляет собой некий логический объект, который обычно реализуется в аппаратуре.  Например,  модуль  может  быть простым вентилем,  32-разрядным счетчиком, подсистемой памяти, вычислительной системой или вычислительной сетью.

     Прежде, чем   обратиться   к   подробностям    языковых конструкций, рассмотрим пример описания модуля (рис. 2.1).

     Модуль add2bit в примере имеет два входа и один  выход. Входы одноразрядные и объявлены проводами (wire). Выход объявлен двухразрядным регистром.  В модуле есть один исполняемый блок,  расположенный между оператором "always ... begin" и "end".  Он постоянно следит за своими входами и когда один из  входов изменяется,  модуль вычисляет значение выхода как сумму двух входов и печатает значение входов, выхода и текущее время моделирования.

,----------------------------------------------------------.

|  module add2bit (in1, in2, sum);                         |

|  input in1, in2;                                         |

|  output[1:0] sum;                                        |

|  wire inl, in2;                                          |

|  reg[1:0] sum;                                           |

|                                                          |

|  always @ (inl or in2)  begin                            |

|     sum = inl + in2;                                     |

|     $display ("The sum of %b and %b is %0d (time = %0d)",|

|     inl, in2, sum, $time);                               |

|  end                                                     |

|  endmodule                                               |

`----------------------------------------------------------'

Рис.2.1 Пример описания модуля

     Хотя модуль  add2bit  очень  простой,  он демонстрирует основные компоненты,  присущие всем модулям. На рис. 2.2 показан синтаксис определения модуля.

     Каждый модуль имеет заголовок (header),  содержащий имя модуля (module name) и список входов и выходов. Он описывает средства, с помощью которых модуль взаимодействует со  своим окружением. Все  (внешние) входы  и  выходы,  так же как все (внутренние) переменные должны быть описаны. Обычно они объявляются либо проводами (wire),  которые просто обеспечивают межсоединения между субблоками модуля, или регистрами, которые могут сохранять информацию и состояние которых может быть изменено функциональными операторами.

,-------------------------------------------------------------.

| module <имя_модуля> <необязательный список входов/выходов>; |

| <описание входов/выходов>                                   |

| <описание локальных переменных>                             |

|                                                             |

| <элемент_модуля>                                            |

|     ...                                                     |

| <элемент_модуля>                                            |

|                                                             |

|  endmodule                                                  |

`-------------------------------------------------------------'

Рис.2.2 Синтаксис определения модуля

     Ядро тела модуля  на  рис.2.3  представлено  <элементами_модуля>.  Эти элементы могут быть различного типа, наиболее распространенными из них являются непрерывное присваивание, структурные элементы и функциональные элементы.

     Непрерывное присваивание - это лаконичный способ описания комбинационной части модуля. Например, строчки

            wire[7:0] result;

            assign result = op1 + op2;

или эквивалентная строчка

            wire[7:0] result = op1 + op2;

определяют result как сумму op1 и op2. Всякий раз, когда op1 или op2 изменяются, result вычисляется заново.

     Структурный элемент - это элемент модуля более  низкого уровня в текущем модуле. Например, строки
            fulladd f1 (cin0, a0, b0, sum0, cout0);

            fulladd f2 (cout0, a1, b1, sum1, cout2);

создают экземпляры двух модулей fulladd в текущем модуле,  и дают им  имена  f1  и  f2.  Структурный элемент эквивалентен встраиванию схемы более низкого уровня в текущую схему. Если модуль состоит только из структурных элементов, тогда модуль представляет собой список элементов.

     Так как структурные элементы только определяют разделение и структуру модуля, должен быть способ описания функциональной сущности модулей самого нижнего уровня.  Один способ описания функции - использовать примитивы языка Verilog, такие  как  вентили AND,  OR и т.д.  Но хотя проектирование на вентильном уровне может быть очень эффективным,  более предпочтителен способ,  который поддерживает Verilog,  заключающийся в использовании функциональных конструкций более высокого уровня для описания функционирования модели, называемых функциональными элементами.

     Функциональным элементом  является  блок  программы  на Verilog,  которому предшествует ключевое слово "initial" или "always".  Все элементы в модуле (как функциональные,  так и структурные) выполняются одновременно.  Функциональный  элемент  "initial" выполняется один раз в начале моделирования. Например, блок

            initial

             i = 0;

             устанавливает i  в  0 в начале моделирования. Функциональный элемент "always" выполняется постоянно в бесконечном  цикле. Например, блок
            always

             #10 i = i + 1;

увеличивает i на 1 каждые 10 единиц времени все время моделирования.

     Остальная часть этой главы  дает обзор  функционального аспекта Verilog.

ОСНОВНЫЕ ТИПЫ ДАННЫХ

     Подобно большинству  языков  программирования,  Verilog имеет константы,  которые хранят фиксированную информацию, и переменные (variables),  которые  могут  изменяться во время моделирования. Константы в Verilog могут  быть  десятичными, шестнадцатиричными, восьмиричными или двоичными и имеют формат

            размер'основание величина

где размер - необязательное  десятичное  целое,  описывающее разрядность  константы,  основание - необязательный параметр, могущий принимать одно из значений b,B,d,D,o,O,h и H.

            В и b обозначают двоичную константу,

            О и о обозначают восьмеричную константу,

            D и d обозначают десятичную константу,

            H и h обозначают шестнадцатеричную константу.

     Основание B и b означает двоичную константу,  основание О и o означает восьмеричную константу, основание D и d означает десятичную константу,  а основание H и h означает шестнадцатеричную константу.

     Если размер  не  указан,  он  вычисляется  из  величины константы,  а если не указано основание,  то подразумевается десятичное основание. Некоторые примеры:

            15        (десятичная 15)

            'h15      (десятичная 21, шестнадцатеричная 15)

            5'b10011  (десятичная 19,двоичная 10011)

            12'h01F   (десятичная 31, шестнадцатеричная 01F)

     Строковые константы   пишутся   в  кавычках  (например, "mystring") и преобразуются в эквивалентный двоичный  формат ASCII. Например, строка "ab" эквивалентна 16'h5758.

     Переменные в  Verilog  могут  иметь  тип reg (регистр), wire (провод),  integer (целое),  real (вещественное), event (событие) и  time  (время).  В  то  время  как типы integer, float, time и event представляют абстрактные  понятия,  типы reg и  wire  непосредственно  отражаются в аппаратуре.  Wire представляет межсоединения между блоками,  reg  представляет элемент памяти.  Как reg, так и wire имеют размер, указанный в описании или равный по умолчанию 1.  Каждый разряд в переменной типа  reg  или  wire  может принимать одно из четырех значений: 1,  0, x или z. х представляет или неинициализированную переменную, или конфликт,  такой как в случае,  когда соединены два выхода и каждый пытается установить  сигнал  в разное состояние. z представляет высокий импеданс или плавающую величину и используется для шин  с  тремя  состояниями. reg и  wire могут использоваться в арифметических выражениях и интерпретируется моделирующей  программой  как  целое  без знака.

     Ниже приведены примеры описания переменных:

       integer i, j;             // два целых числа

       real f, d;                // два вещественных числа

       wire [7:0] bus;           // 8-и разрядная переменная

                                 // bus (шина)

       reg [0:15] word;          // 16-и разрядная переменная

                                 // word (слово)

       event trigger,clock_high; // два события
       time t_setup,t_hold;      // t1, t2

     C одноразрядными и многоразрядными подполями переменных типа reg  и wire действуют как с другими переменными и к ним можно обращаться и изменять в выражениях:

        reg [15:0] word;

        reg [7:0] byte;

        ...

        word[0] = ...            // первый справа разряд

                                 // переменной word

        byte = ... +word[15:8];  //  левая половина word

        ... word ...             //  эквивалентно  word[15:0]

     Следующие примеры иллюстрируют использование переменных в выражениях Verilog:

        i = i + j;

        assign bus = word[15:8] + word[7:0];

        f = (g + 1.2) * 3.29E-5;

        @e1 #5 ->e2;  // ждать совершения события е1

                      // ждать 5 единиц времени

                      // и запустить событие е2

        if ($time - t1 < t2) error ("timing violations");

     Целые - это 32-разрядные числа со знаком, которые могут сохранять рабочие переменные. Переменные типа time могут содержать значение  модельного  времени и обычно реализуются в виде 64-разрядных целых без знака.  Переменные типа event  и time будут  рассмотрены  позже  в разделе "Понятие времени и событий".

     В Verilog  есть только одна структура данных,  а именно array (массив),  которая может иметь тип  любой  переменной. Далее следуют примеры описания массива:

      integer num[99:0];      // массив из 100 целых

      reg [7:0]mem[0:1024];   // массив из 1024 байтов

      reg [10*8:1]names[20];  // массив из 20 переменных

                              // names, каждая имеет длину

                              // 10 символов, каждый символ

                              // является 8-разрядным байтом

     Обращение к  элементу массива имеет такой же синтаксис, как обращение к разряду вектора, таким образом word[i] может быть  i-ым элементом в массиве word или i-ым разрядом в векторе word,  в зависимости от того, как описано word. Для обращения  к разряду подполя или элемента массива элемент сначала надо записать в рабочую переменную:

       reg [15:0] array [0:10];

       reg [15:0]temp;

       ...

       temp = array[3];

       ...temp[7:5]

ОСНОВНЫЕ ОПЕРАЦИИ И ВЫРАЖЕНИЯ

     Язык Verilog  заимствует  синтаксис  и  семантику большинства операций из языка программирования  Си.  Исключением является отсутствие  операций  увеличение  (++) и уменьшение(--). На рис.2.3  приводится  сводка  операций  Verilog.  На рис.2.6 приводится порядок выполнения операций.

,----------------------------------------------------.

|      +  -   *  /        (арифметические операции)  |

|      >  >=  <  <=       (операции отношения)       |

|      !  &&  ||          (логические операции)      |

|      == !=              (логическое равенство)     |

|      ?:                 (условная)                 |

|      { }                (конкатенация)             |

|      %                  (операция "по модулю")     |

|      === !==            (сравнение)                |

|      ~  &  |            (по-битовые)               |

|      <<  >>             (сдвиг)                    |

`----------------------------------------------------'

Рис.2.3 Операции языка

     Все операции над двумя операндами можно также использовать как одноместные редукционные операции.  Например, выражение "+varname" даст сумму всех разрядов varname. Более типичный пример редукционного  использования  "^varname",  что дает исключительное  ИЛИ всех разрядов varname,  то есть его четность. Другие примеры одноместных  редукционных  операций представляет выражение

      ^word === 'bx // истинно,  если в  word  есть  разряд,

                    // равный х

или выражение

      &word == 0    // истинно, если в word есть разряд,

                    //  равный 0

-----------.       Отличие логических ||, &&  от  |, &

Примечание |      -------------------------------------

Шевцова С. |  1) при &,| неравные векторы выравниваются по 

-----------'     большему, дополняя нулями слева меньший, и

                 после этого производится поразрядное срав-

                 нение векторов по & или |. Результат - век-

                 тор поразрядных сравнений.

              2) при &&, || вектора сначала оцениваются: ес-

                 ли в векторе есть хотя бы одна единица, он

                 считается равным 1, иначе - равным 0, - и

                 после оценки выполняется операция & или |

                 уже над однобитовыми значениями; результат -

                 также  однобитовый.

     Операции логического сравнения идентичны  соответствующим языка Си. При сравнении двух переменных, которые не равны 0 или 1 ( то есть х или z) операция эквивалентности  дает результат ложь.  Для проверки равенства переменных,  которые могут содержать х или z,  Verilog имеет  операции  сравнения "===" и  "!==".  На  рис.2.4  показана  разница между "==" и "===".

,-------------------------------------------------------.

|      module equ_equ;                                  |

|                                                       |

|        initial begin                                  |

|          $display ("'bx == 'bx is %b", 'bx == 'bx);   |

|          $display ("'bx === 'bx is %b", 'bx === 'bx); |

|          $display ("'bz != 'bx is %b", 'bz != 'bx);   |

|          $display ("'bz !== 'bx is %b", 'bz!== 'bx);  |

|        end                                            |

|                                                       |

|      endmodule                                        |

`-------------------------------------------------------'

Рис.2.4 Различие между " == " и " ==="

     Выполнение модуля equ_equ даст следующие результаты:

           'bx == 'bx is x

           'bx === 'bx is 1

           'bz != 'bx is x

           'bz !== 'bx is 1

     Две операции,  отсутствующие  в Си,  это конкатенация и тиражирование. Две и более переменные (или константы)  могут быть конкатенированы  заключением  их в фигурные скобки { }, разделенные запятой:

           {2'b1x, 4'h7} === 6'b1x0111

     Конкатенированные переменные  могут содержаться в любом выражении или в левой части оператора присваивания. Константа, участвующая  в конкатенации,  должна иметь явно заданный размер ( то есть 1'bz , а не 'bz). Размер конкатенации равен сумме размеров ее составляющих.

     Тиражирование обеспечивает  краткую  форму  записи  для дублирования константы или переменной.  Выражение может быть тиражировано заключением его в двойные фигурные скобки с помещением коэффициента тиражирования между двумя открывающими скобками:

           {3{2'b01}} === 6'b010101

     На рис.2.5 показан  пример  использования  конкатенации для перестановки  двух  байтов и использование тиражирования для знакового расширения переменной word.

,----------------------------------------------.

|   module concat_replicate(swap,signextend);  |

|                                              |

|       input swap,signextend;                 |

|                                              |

|       reg[15:0] word;                        |

|       reg[31:0] double;                      |

|       reg[7:0] byte1, byte2;                 |

|                                              |

|       initial begin                          |

|          byte1 = 5; byte2 = 7;               |

|          if(swap)                            |

|             word = {byte2, byte1};           |

|          else                                |

|             word = {byte1, byte2};           |

|          if(signextend)                      |

|             double = {{16{word[15]}},word};  |

|          else                                |

|             double = word;                   |

|       end                                    |

|                                              |

|   endmodule                                  |

`----------------------------------------------'

Рис.2.5 Конкатенация и тиражирование

     Выражения можно переставлять, используя обычные правила последовательности выполнения операций.  Можно  использовать скобки для улучшения читаемости и избежания двусмысленности.

ПРОЦЕДУРНЫЕ ОПЕРАТОРЫ

     Хотя выражения  можно использовать для вычисления величины, их  нельзя  вычислять  изолировано,  они  должны  быть частью оператора.  Простой  оператор  может  быть оператором присваивания или оператором  порядка  выполнения.  Составной оператор или блок состоит из группы операторов,  заключенных в "begin" и "end".  Каждый функциональный элемент состоит из простого или составного оператора. Блок может быть именован:

              begin : instruction_fetch

              ...

              end

где идентификатор,  следующий за ":" - это имя блока. Именованные блоки могут иметь внутри себя описание локальных  переменных.

,---------------------------------------------------.

|         binary op.        самый высокий приоритет |

|         !~                                        |

|         *  /  %                                   |

|         +  -                                      |

|         <<  >>                                    |

|         <  <=  >  >=                              |

|         ==  !=  === !==                           |

|         &                                         |

|         ^  ^~                                     |

|         |                                         |

|         &&                                        |

|         ||                                        |

|         ?:                 самый низкий приоритет |

`---------------------------------------------------'

Рис.2.6 Приоритет операций

     Язык Verilog  имеет  много  управляющих  конструкций из других языков высокого уровня, особенно из языка Си.

Цикл for

     Cледующий пример  (рис.2.7)  показывает   использование цикла for.

     Выполнение модуля for_loop дает следующие результаты:

              i = 0 (0 в двоичном виде)

              i = 1 (1 в двоичном виде)

              i = 2 (10 в двоичном виде)

              i = 3 (11 в двоичном виде)

,---------------------------------------------------.

|   module for_loop                                 |

|                                                   |

|       integer i;                                  |

|                                                   |

|       initial                                     |

|          for (i = 0; i < 4; i = i + 1) begin      |

|          $display ("i = %0d (%b binary)", i, i);  |

|       end                                         |

|                                                   |

|   endmodule                                       |

`---------------------------------------------------'

Рис.2.7 Оператор цикла for

Цикл c условием продолжения while

     Результат выполнения цикла for можно получить использованием конструкции while,  как показано на рис.2.8. Выполнение модуля while loop дает такие же результаты, как и выполнение модуля for loop на рис. 2.7.

Оператор выбора

     Следующий пример (рис.2.9) показывает использование управляющей структуры  выбора.  Оператор  выбора соответствует переключателю  switch языка Си.

,--------------------------------------------------.

|   module while_loop                              |

|       integer i;                                 |

|       initial begin                              |

|         i = 0;                                   |

|         while (i < 4) begin                      |

|           $display ("i = %0d (%b binary)", i, i);|

|          i = i + 1;                              |

|         end                                      |

|       end                                        |

|   endmodule                                      |

`--------------------------------------------------'

Рис.2.8 Оператор цикла while

,--------------------------------------.

|   module case_statement;             |

|        integer i;                    |

|        initial i = 0;                |

|        always begin                  |

|          $display ("i = %0d", i);    |

|          case (i)                    |

|            0: i = i + 2;             |

|            1: i = i + 7;             |

|            2: i = i - 1;             |

|            default: $stop;           |

|          endcase                     |

|        end                           |

|   endmodule                          |

`--------------------------------------'

Рис.2.9 Оператор выбора

     Выполнение модуля case statement дает следующие результаты:

              i = 0

              i = 2

              i = 1

              i = 8

     Выбирающее выражение (в скобках) сравнивается со значениями (перед ":"), совпадение будет при поразрядном совпадении (подобно оператору ===).  Если нет совпадения ни с одним вариантом, выполняется действие по  несовпадению  (default).

Если действие  по  несовпадению не описано,  то продолжается выполнение после оператора выбора.  Хорошим  программистским тоном является  всегда  описывать действие при несовпадении. Если такое не может случиться,  то  модель  должна  печатать сообщение об ошибке или прекратить моделирование.

     Операторы casez и casex очень похожи на оператор  case, за одним исключением. Оператор casez не обращает внимание на разряды, находящиеся в состоянии z,  а оператор casex не обращает внимание на разряды, находящиеся в состоянии z и x.

     Цикл repeat

     В языке Verilog есть еще две управляющие структуры,  не являющиеся обычными  для других языков программирования.  На рис.2.10 показан цикл repeat, который ждет 5 тактов, а затем прекращает моделирование.

     Цикл forever

     На рис.2.11 показан цикл forever, который следит за некоторым условием и выдает сообщение,  когда условие выполняется.

     Хотя операторы repeat и forever можно реализовать с помощью других  управляющих  операторов,  например,  с помощью оператора for,  они очень удобны, особенно при выдаче команд с клавиатуры.  Их  преимущество  в  том,  что они не требуют описания переменных заранее.

,------------------------------------.

|   module repeat_loop (clock);      |

|       input clock;                 |

|                                    |

|       initial begin                |

|          repeat (5)                |

|              @ (posedge clock);    |

|              $stop;                |

|          end                       |

|                                    |

|   endmodule                        |

`------------------------------------'

Рис.2.10 Цикл repeat

,----------------------------------------------------.

|   module forever_statement (a,b,c);                |

|       input a,b,c;                                 |

|                                                    |

|       initial forever begin                        |

|         @ (a or b or c)                            |

|         if (a + b == c) begin                      |

|            $display ("a(%d)+b(%d) = c(%d)", a,b,c);|

|            $stop;                                  |

|         end                                        |

|       end                                          |

|                                                    |

|   endmodule                                        |

`----------------------------------------------------'

Рис.2.11 Цикл forever

ПОНЯТИЕ ВРЕМЕНИ И СОБЫТИЙ

     До сих  пор  описание  поведения на языке Verilog очень похоже на программирование на любом структурированном  языке высокого уровня. Единственное и наиболее важное различие – в понятии времени и его влиянии на порядок выполнения операторов в модуле.  В большинстве языков программирования существует единственный программный счетчик,  указывающий  текущее место выполнения программы. Так как в аппаратуре все элементы работают параллельно,  последовательная модель выполнения не соответствует языку описания аппаратуры и, таким образом, выполнение в Verilog управляется событиями. Глобальная переменная  обозначает модельное время.  В каждый момент времени может быть одно или более событий,  которые  следует  выполнить.  Планировщик  событий  программы моделирования Verilog simulator занимает место программного счетчика языков  программирования.

     Verilog simulator выполняет все события,  запланированные на текущее модельное время и удаляет их из списка  событий.  Когда  в  текущее  модельное время нет больше событий, тогда модельное время продвигается к первому элементу,  запланированному  на следующий раз.  По мере выполнения событий генерируются новые события для будущего времени  (или,  возможно, для текущего). На рис.2.12 показана ось модельного времени с  несколькими событиями,  запланированными в различных точках.  Заметим,  что время перед текущим модельным  временем  не  может иметь никаких событий,  связанных с ним, так как все события были выполнены и удалены.

     Порядок выполнения  событий  внутри  одного и то же модельного времени, в общем случае, не известен и нельзя полагаться на  него,  так как Verilog simulator может попытаться оптимизировать выполнение упорядочиванием событий особым образом. Однако,  Verilog гарантирует, что линейный код, который не имеет управления временем, будет выполняться как одно событие без прерывания. Также гарантируется, что порядок выполнения двух одинаковых прогонов моделирования будет одинаковым. Пример  на  рис.2.13  иллюстрирует это использованием нескольких функциональных элементов.

      текущее   t1               t2               t3

     -----------,----------------,----------------,------->

      время     |                |                |

          ,-----'-----.    ,-----'-----.    ,-----'-----.

          | событие 0 |    | событие 3 |    | событие 4 |

          `-----,-----'    `-----------'    `-----,-----'

          ,-----'-----.                     ,-----'-----.

          | событие 1 |                     | событие 5 |

          `-----,-----'                     `-----------'

          ,-----'-----.

          | событие 2 |

          `-----------'

            Рис.2.12 Ось времени

,--------------------------------------------------------.

|   module event_control;                                |

|                                                        |

|     register [4:0] r;                                  |

|                                                        |

|    initial begin                                       |

|        $display ("First initial block, line 1.");      |

|        $display ("First initial block, line 2.");      |

|    end                                                 |

|                                                        |

|    initial                                             |

|       for (r = 0; r <= 3; r =r + 1)                    |

|           $display ("r = %0b",r);                      |

|                                                        |

|   endmodule                                            |

`--------------------------------------------------------'

Рис.2.13 Несколько функциональных элементов

     Выполнение модуля  event_control дает следующие результаты:

        Fist initial block, line 1.

        Fist initial block, line 2.

        r = 0

        r = 1

        r = 10

        r = 11

     Здесь видно,  что все блоки initial по плану должны выполняться в одно и то же модельное время (момент 0).

     Обращаясь к результатам моделирования на рис.2.13, можно видеть, что Verilog выбирает некоторый порядок выполнения блоков,  который не очевиден из рассмотрения программы модели. Заметим, однако, что если блок был запланирован к выполнению, он продолжает выполняться до полного окончания.

УПРАВЛЕНИЕ ВРЕМЕНЕМ И СОБЫТИЯМИ

     Процесс языка Verilog (то есть блок initial или always) может перепланировать  свое собственное выполнение использованием одной из трех форм управления временем:

       #выражение

       @событие-выражение

       wait (выражение)

     Форма #выражение, используемая для синхронного управления, останавливает выполнение процесса на фиксированное время, указанное  параметром "время",  а форма @событие-выражение, используемая для асинхронного контроля ,  останавливает выполнение до тех пор, пока не случится указанное событие. В обоих случаях планировщик удаляет  исполняемые  в  настоящий момент события из списка событий текущего модельного времени и заносит их в некоторый список будущих событий.

     Форма wait  (выражение)  -  это  управление  событиями, чувствительное к уровню. Если wait выражение ложно, выполнение  будет  остановлено  до  тех  пор,  пока  оно не  станет истинным (из-за выполнения  некоторого  оператора  в  другом процессе).

     Пример на рис.2.14 показывает  использование  структуры управления временем.

     Выполнение модуля time_control дает следующие результаты:

        r = 2 в момент 5

        r = 1 в момент 10

        r = 2 в момент 25

        r = 1 в момент 30

        r = 2 в момент 55

        r = 1 в момент 60

     Заметим, что первый оператор initial

          initial #70 $stop;

- это обычный способ для остановки моделирования в определенный момент времени.

     В этом  примере,  хотя  все  блоки initial начинаются в один и тот же момент времени, некоторые из них были остановлены (и  перепланированы) в разные моменты модельного времени. Обратите внимание на  использование  именованных  блоков (b1 и b2). Именованный блок может внутри себя иметь описание локальных переменных,  хотя в этом примере это  свойство  не используется, а имена используют только для удобства записи.

     Форма управления @событие-выражение ждет,  чтобы случилось событие, прежде, чем продолжать выполнение блока. Событие может быть одной из нескольких форм:

     а) переменная <или переменная> ....

     b) posedge одноразрядная переменная

     c) negedge одноразрядная переменная

     d) событие-переменная

     В форме  а)  выполнение задерживается до тех пор,  пока одна из переменных не изменится.  В форме b) и с) выполнение задерживается, пока переменная не изменится от 0,  х или z к

,-----------------------------------------------------.

|    module time_control;                             |

|                                                     |

|    reg[1:0] r;                                      |

|                                                     |

|    initial #70 $stop;                               |

|                                                     |

|    initial begin : b1 // именованный блок b1        |

|       #10 r = 1; // ждать 10 единиц времени         |

|       #20 r = 1; // ждать 20 единиц времени         |

|       #30 r = 1; // ждать 30 единиц времени         |

|    end                                              |

|                                                     |

|    initial begin : b2 // именованный блок b2        |

|       #5 r = 2; // ждать 5 единиц времени           |

|       #20 r = 2; // ждать 20 единиц времени         |

|       #30 r = 2; // ждать 30 единиц времени         |

|    end                                              |

|                                                     |

|     always @r begin                                 |

|       $display ("r = %0d at time %0d", r, $time);   |

|     end                                             |

|                                                     |

| endmodule                                           |

`-----------------------------------------------------'

Рис.2.14 Пример управления временем

1 (в случае posedge) или от  1,  х  или  z  к  0  (в  случае negedge). В форме d) выполнение блока останавливается до тех пор, пока не выполнится событие.

,-------------------------------------------------------.

|    module event_control;                              |

|                                                       |

|       event e1,e2;                                    |

|                                                       |

|      initial @e1 begin                                |

|         $display ("I am in the middle.");             |

|         ->e2;                                         |

|      end                                              |

|                                                       |

|      initial @e2                                      |

|        $display ("I am supposed to execute last.");   |

|                                                       |

|      initial begin                                    |

|         $display ("I am the first.");                 |

|         ->e1;                                         |

|      end                                              |

|                                                       |

|   endmodule                                           |

`-------------------------------------------------------'

Рис.2.15 Пример управления событиями

     Событие может  быть  запущено   выполнением   выражения->cобытие-переменная. Следующий пример на рис.2.15 использует событие-переменные для  управления  порядком   выполнения трех блоков initial, которые выполняются в один и тот же момент модельного времени.

     Выполнение модуля  event_control даст следующие результаты:

     I am the first.

     I am in the middle.

     I am supposed to execute last.

     Эта форма управления обеспечивает  порядок  выполнения. Без оператора управления событиями планировщик языка Verilog может спланировать выполнение блоков initial в  произвольном порядке.

     Особой формой конструкции управления временем и событиями является ее использование внутри оператора присваивания.

Присваивание current_state = #clock_period next_state;

эквивалентно следующим двум операторам

     temp = next_state;

     #clock_period current_state = next_state;

и, аналогично, присваивание

     current_state = @ (posedge clock) next_state;

эквивалентно двум операторам

     temp = next_state;

     @ (posedge clock) current_state = temp;

ПОНЯТИЕ ПАРАЛЛЕЛИЗМА

     Язык Verilog имеет еще несколько управляющих  структур, которые  необычны  для других языков программирования.  Одна из них - конструкция ветвление-соединение (fork-join),  другая - оператор блокировки (disable statement).

     Пара fork-join

     На рис.2.16  показан  пример  использования конструкции ветвление-соединение.

     В этом примере выполнение блока initial будет  остановлено,  пока не будут выполнены в некотором порядке события а и b.  Во время ветвления (fork) активизируется два или более путей выполнения. Когда все будут закончены, продолжается выполнение в соединении (join). Если некоторые из путей заканчиваются раньше  других,  эти  пути  останавливаются  и ждут остальные.

,------------------------------------------.

|   module fork_join;                      |

|                                          |

|      event a, b;                         |

|                                          |

|      initial                             |

|        fork                              |

|          @a;                             |

|          @b;                             |

|        join                              |

|      end                                 |

|                                          |

|   endmodule                              |

`------------------------------------------'

Рис.2.16 Пример параллельных процессов

     Оператор блокировки

     Оператор блокировки работает, как оператор  прекращения break языка Си. Но в то время как оператор break в программе на языке Си только изменяет  программный  счетчик,  оператор disable должен  удалить  ждущие события из очередей событий. Оператор блокировки берет имя блока в качестве  аргумента  и удаляет остальные события,  связанные с этим блоком, из очереди. Можно блокировать только именованные блоки и задачи. Следующий пример на рис.2.17 - это модификация предыдущего, но вместо ожидания выполнения обоих  событий  а  и  b, этот модуль ждет только одного из них - или а, или b.

,------------------------------------------.

|   module disable_block;                  |

|      event a, b;                         |

|      // Требуется имя блока              |

|      initial   begin : block1            |

|        fork                              |

|          @a disable block1;              |

|          @b disable block1;              |

|        join                              |

|      end                                 |

|   endmodule                              |

`------------------------------------------'

Рис.2.17 Пример блокировки

ФУНКЦИИ И ЗАДАЧИ

     Один из  наиболее  мощных методов моделирования в языке Verilog - инкапсуляция части программы в виде задачи  (task) или функции (function). На рис.2.19 показан пример задачи.

,-------------------------------------------------.

|   function [7:0] func;                          |

|      input i1;                                  |

|      integer i1;                                |

|                                                 |

|      reg [7:0] rg;                              |

|                                                 |

|      begin                                      |

|       rg=1;                                     |

|       for (i=1; i<=i1; i++)                     |

|          rg=rg+1;                               |

|       func=rg;                                  |

|      end                                        |

|   endfunction                                   |

`-------------------------------------------------'

Рис.2.18 Пример функции

,-----------------------------------------------------.

| task tsk;                                           |

|                                                     |

|      input  i1,i2;                                  |

|      output o1,o2;                                  |

|                                                     |

|       $display ("Task tsk, i1=%0b, i2=%0b", i1,i2); |

|       #1 o1 = i1 & i2;                              |

|       #1 o2 = i1 | i2;                              |

|                                                     |

| endtask                                             |

`-----------------------------------------------------'

Рис.2.19 Пример задачи

     Есть несколько различий между задачей и функцией.

     Задача может содержать конструкции управления временем, а функция не может. Это означает, что функция выполняется за нулевое модельное  время  и  выдает результат немедленно (то есть она по существу комбинационная), а задача  может  иметь задержку и программа, запускающая задачу, должна ждать, пока не закончится выполнение задачи или задача не будет блокирована,  прежде, чем продолжить выполнение программы. Управление выполнением возвращается  к  оператору,  непосредственно следующему за оператором,  который запустил задачу или функцию.

     Задача может иметь как входы,  так и выходы,  а функция должна иметь  по  меньшей мере один вход и не имеет выходов. Функция выдает результат по ее имени.

     Вызов задачи - это оператор, а функция вызывается, когда на нее ссылаются в выражении. Например,

     tsk (out, in1, in2);

вызывает задачу с именем tsk, а

     i = func (a, b, c); // или
     assign x = func (y);

вызывает функцию с именем func.

     На рис.2.18 показан пример функции. Функции играют  важную  роль в логическом синтезе.  Так как функции являются комбинационными, их можно синтезировать и использовать в описании систем.  Задачи - это очень важный инструмент организации программы и улучшения ее читаемости и простоты сопровождения.  Часть программы,  которая используется более одного раза,  следует инкапсулировать  в  задачу. Это  помогает  локализовать  любое  изменение  до этой части программы.  Если ожидается,  что программа будет  выдаваться интерактивно  с терминала,  ее также следует преобразовать в задачу для того, чтобы сэкономить на вводе. Кроме того, полезно разбивать длинные процедурные блоки на меньшие задачи, чтобы улучшить читаемость программы.

ФУНКЦИОНАЛЬНОЕ ОПИСАНИЕ

     Verilog - это язык для проектирования сверху вниз,  который обеспечивает  функциональное   описание,   структурное описание и смешанное описание. В следующих нескольких разделах мы проиллюстрируем эти аспекты языка на примере  полного описания 4-разрядного сумматора.

     Первый пример на рис.2.20  описывает  поведение  модуля 4-разрядного сумматора с использованием конструкций высокого уровня языка Verilog.

     Заметим, что порты sum и  zero были объявлены регистрами. Это сделано,  чтобы функциональный оператор мог присваивать им  значение.  В  этом описании есть два функциональных элемента, один элемент initial и один элемент always.

     Конструкция @(in1  или  in2)  заставляет  моделирование ждать, пока один из входов  in1  или  in2  не  изменит  свое последнее значение.  Без этой конструкции цикл always всегда бы выполнялся с одними и теми же значениями входов и модельное время никогда не продвинется.

     Заметим, блок always можно  также запрограммировать как блок initial,  используя  управляющую структуру forever, как показано на рис.2.21.

,----------------------------------------------------.

|  module adder4 (in1,in2,sum,zero);                 |

|                                                    |

|       input [3:0] in1;                             |

|       input [3:0] in2;                             |

|       output [4:0] sum;                            |

|       output zero;                                 |

|       reg [4:0] sum;                               |

|       reg zero;                                    |

|                                                    |

|       initial begin                                |

|         sum = 0;                                   |

|         zero = 1;                                  |

|       end                                          |

|                                                    |

|       always @ (in1 or in2) begin                  |

|          sum = in1 + in2;                          |

|          if (sum == 0)                             |

|             zero = 1;                              |

|          else                                      |

|             zero = 0;                              |

|       end                                          |

|    endmodule                                       |

`----------------------------------------------------'

Рис.2.20 Функциональное описание 4-х разрядного сумматора

     Следующий пример на рис.2.22 является модификацией примера на рис.2.20,  он моделирует adder4, используя непрерывное присваивание.

     Здесь zero является проводом, а не регистром. Когда изменяется sum после регистра, zero перевычисляется, используя третичную операцию "?:" ,  которая имеет тот же смысл, что и в языке Си, и является выражением,  эквивалентным  оператору если-то-иначе. Объявление  zero  проводом  и его непрерывное присваивание можно записать в один оператор следующим  образом:

     wire zero = (sum == 0) ? 1 : 0;
,----------------------------------------------.

|    initial begin                             |

|       forever begin                          |

|        @ (in1 or in2) begin                  |

|           sum = in1 + in2;                   |

|           if (sum == 0)                      |

|              zero = 1;                       |

|           else                               |

|              zero = 0;                       |

|        end                                   |

|       end                                    |

|    end                                       |

`----------------------------------------------'
Рис.2.21 Использование initial-forever вместо always
,-----------------------------------------------.

|   module adder4 (in1,in2,sum,zero);           |

|        input [3:0] in1;                       |

|        input [3:0] in2;                       |

|        output [4:0] sum;                      |

|        reg [4:0] sum;                         |

|        output zero;                           |

|                                               |

|        assign zero = (sum == 0) ? 1 : 0;      |

|                                               |

|        initial sum = 0;                       |

|                                               |

|        always @ (in1 or in2)                  |

|           sum = in1 + in2;                    |

|   endmodule                                   |

`-----------------------------------------------'

Рис.2.22 Использование непрерывного присваивания

СТРУКТУРНОЕ ОПИСАНИЕ

     В примере на рис.2.23 показана реализация модуля 4-х разрядного сумматора как  структуры,  состоящей  из  субмодулей 1-разрядного полного сумматора и вентилей.

,------------------------------------------------------.

|   module adder4 (in1,in2,sum,zero);                  |

|                                                      |

|        input [3:0] in1;                              |

|        input [3:0] in2;                              |

|        output [4:0] sum;                             |

|        output zero;                                  |

|                                                      |

|       fulladd u1 (in1[0],in2[0], 0,sum[0],c0);       |

|       fulladd u2 (in1[1],in2[1],c0,sum[1],c1);       |

|       fulladd u3 (in1[2],in2[2],c1,sum[2],c2);       |

|       fulladd u4 (in1[3],in2[3],c2,sum[3],sum[4]);   |

|                                                      |

|    nor u5 (zero,sum[0],sum[1],sum[2],sum[3],sum[4]); |

|                                                      |

|   endmodule                                          |

`------------------------------------------------------'

Рис.2.23 Структурное описание 4-х разрядного сумматора

     В этом примере 4-х разрядный сумматор состоит из  четырех элементов модулей  1-разрядного сумматора (fulladd) и одного элемента модуля вентиля НЕ-ИЛИ.  В этой реализации  описывается  аппаратная  структура  и есть однозначное соответствие со схемотехникой. Она использует два типа модулей более низкого уровня:  fulladd  и nor.  Хотя модуль fulladd определен пользователем, а nor  -  это  примитив  языка  Verilog,  оба используются одинаковым способом.

     Заметим, что структурное описание  неявно объявляет три провода: с0,  с1 и с2. Эти провода связывают разряд переноса одной  ступени fulladd со  входом  другой  ступени.  Verilog разрешает неявное  объявление  одиночных разрядов проводами. Если межсоединение представляет собой  многоразрядную  шину, требуется явное описание, например:

     wire [3:0] databus;

     Заметим также, что когда используется модуль, важен порядок входных/выходных портов. Модуль более  высокого уровня adder4  требует  установления связи между его схемой и соответствующими портами модуля более низкого уровня fulladd. На рис.2.24 показана простая реализация модуля fulladd.

,------------------------------------------------------.

|   module fulladd (in1,in2,сarryin,sum,carryout);     |

|                                                      |

|        input  in1,in2,carryin;                       |

|        output sum,carryout;                          |

|                                                      |

|        assign {carryout,sum} = in1 + in2 + carryin;  |

|   endmodule                                          |

`------------------------------------------------------'

Рис.2.24 Функциональное описание 1-разрядного полного сумматора

СМЕШАННОЕ ПРЕДСТАВЛЕНИЕ

     Последний пример на рис.2.25 показывает adder4 как комбинацию структурных и функциональных элементов.

     Эта модель вычисляет выход sum посредством  структурных элементов модулей   fulladd,  а  выход  zero  вычисляется  с использованием функционального элемента.

,------------------------------------------------------.

|   module adder4 (in1,in2,sum,zero);                  |

|                                                      |

|        input [3:0] in1;                              |

|        input [3:0] in2;                              |

|        output [4:0] sum;                             |

|        output zero;                                  |

|        reg zero;                                     |

|                                                      |

|       fulladd u1 (in1[0],in2[0], 0,sum[0],c0);       |

|       fulladd u2 (in1[1],in2[1],c0,sum[1],c1);       |

|       fulladd u3 (in1[2],in2[2],c1,sum[2],c2);       |

|       fulladd u4 (in1[3],in2[3],c2,sum[3],sum[4]);   |

|                                                      |

|       always @ sum                                   |

|         if (sum == 0)                                |

|           zero = 1;                                  |

|         else                                         |

|           zero = 0;                                  |

|                                                      |

|   endmodule                                          |

`------------------------------------------------------'

Рис.2.25 Смешанное представление

Глава 3. МОДЕЛИРОВАНИЕ КОНВЕЙЕРНОГО ПРОЦЕССОРА

     В этой главе мы шаг за  шагом  построим  функциональную модель  32-х разрядного процессора.  Сначала мы создадим модель системы команд,  затем опишем уровень регистровых передач.  В  следующей главе мы приступим к описанию структурной модели,  в которой представим процессор как совокупность отдельных блоков.  При этом мы объясним, как моделировать конвейерную обработку, параллелизм, выполнение команд, функциональное разбиение и создание тестовых векторов.

     Основное внимание уделено процессу моделирования,  а не описанию  архитектуры процессора.  Мы не собираемся подробно объяснять функционирование коммерческого микропроцессора или архитектуру.  Некоторое  описание  архитектуры дано для объяснения понятий и процесса моделирования.

ПРИМЕР SISC ПРОЦЕССОРА

     Обычный процессор на СБИС определяется своей архитектурой и системой команд. Давайте определим ЭВМ с малым набором команд (SISC) как компьютер, который выполняет только 10 команд:  считывание из памяти (load), запись в память (store), сложение    (add),    умножение    (multiply),    дополнение (complement),  сдвиг  (shift),  циклический  сдвиг (rotate), пустая команда (nop),  команда останова (halt) и команда перехода (branch).  Мы будем проектировать процессор,  который может выполнять этот малый набор команд.

     Перед обсуждением  модели  процессора мы должны понять, как он выполняет программы,  содержащие смесь  этих  команд. Именно это мы собираемся определить с помощью модели системы команд.

     Структурная схема  SISC  системы  представлена  на рис. 3.1, а система команд - на рис.3.2.

  ,-------------.

  |  Генератор  |         ,--------------------------------.

  |   тактовых  |---------|           ПРОЦЕССОР            |

  |   импульсов |         |                                |

  `-------------'         |  ,----------.  ,-------------. |

                          |  |   РС     |  |   PSR       | |

                          |  | (Счетчик |  |  (Регистр   | |

                          |  |  команд) |  | состояние   | |

  ,-------------.         |  |          |  |  процессора)| |

  |             |         |  `----------'  `-------------' |

  |             |  Адрес  |                                |

  |             |<------/-|      ,-----------------.       |

  |             |         |      |       IR        |       |

  |    Память   |  Данные |      | (Регистр команд)|       |

  |             |------/->|      `-----------------'       |

  |             |<----/---|                                |

  |             |         |                                |

  |             |         |                                |

  |             |         |  ,---------------------------. |

  `-------------'         |  |           ALU             | |

                          |  |          (АЛУ)            | |

                          |  `---------------------------' |

                          |                                |

                          |                                |

                          |  ,---------------------------. |

                          |  |      Register File        | |

                          |  |    (Регистровый файл)     | |

                          |  |                           | |

                          |  `---------------------------' |

                          `--------------------------------'

     Рис. 3.1 Пример структурной схемы SISC процессора

,-------------------------------------------------------.

|  Команды                                              |

|                                                       |

|  Имя       Обозначение   Код     Формат (inst dst scr)|

|                        операции                       |

|                                                       |

|  NOP            NOP       0       NOP                 |

|  BRANCH         BRA       1       BRA    mem, cc      |

|  LOAD           LD        2       LD     reg, mem1    |

|  STORE          STR       3       STR    mem, srс     |

|  ADD            ADD       4       ADD    reg, srс     |

|  MULTIPLY       MUL       5       MLT    reg, srс     |

|  COMPLEMENT     CMP       6       CMP    reg, srс     |

|  SHIFT          SHF       7       SHF    reg, cnt     |

|  ROTATE         ROT       8       ROT    reg, cnt     |

|  HALT           HLT       9       HLT                 |

|                                                       |

|  Коды условий                                         |

|                                                       |

|  A      Always   (всегда)              0              |

|  C      Carry    (перенос)             1              |

|  E      Even     (четный результат)    2              |

|  P      Parity   (четность результата) 3              |

|  Z      Zero     (нуль)                4              |

|  N      Negative (отрицательный)       5              |

|                                                       |

|  Адресация операндов                                  |

|                                                       |

|  mem - адрес в памяти                                 |

|  mem1 - адрес в памяти или непосредственный операнд   |

|  reg - индекс  регистра                               |

|  scr - индекс регистра или непосредственный операнд   |

|  cc - код условий                                     |

|  cnt - величина сдвига в командах shift/rotate,       |

|                >0=вправо, <0=влево                    |

|                                                       |

|  Формат команды                                       |

|                                                       |

|  IR[31:28]    код операции                            |

|  IR[27:24]    коды условий                            |

|  IR[27]       тип источника 0=reg(mem), 1=непосред-   |

|               ственный операнд (imm)                  |

|  IR[26]       тип назначения 0=reg, 1=imm             |

|  IR[23:12]    адрес источника                         |

|  IR[23:12]    величина сдвига в командах shift/rotate |

|  IR[11:0]     адрес назначения                        |

|                                                       |

|  Регистр состояния процессора (PSR)                   |

|                                                       |

|  PSR[0]     Carry     (перенос)                       |

|  PSR[1]     Even      (четный результат)              |

|  PSR[2]     Parity    (четность результата)           |

|  PSR[3]     Zero      (ноль)                          |

|  PSR[4]     Negative  (меньше нуля)                   |

`-------------------------------------------------------'

Рис. 3.2 Система команд SISC процессора

МОДЕЛЬ СИСТЕМЫ КОМАНД

     Модель системы  команд  процессора описывает выполнение команд и взаимодействие между ними. При этом не рассматривается  аппаратная реализация этих команд.  Например,  команда сложения (add) может быть промоделирована просто как

     {carry, sum}=in1+in2;

без уточнения деталей,  а именно: было ли сложение выполнено с помощью сумматора со сквозным переносом, сумматора с ускоренным  переносом  или  с использованием какого-либо другого алгоритма; в вычислении суммы и переноса при сложении участвуют  только  два входных сигнала (in1 и in2),  как показано выше.

     Мы также  не будем интересоваться реализацией протокола памяти.  Память будет рассматриваться как большой набор  регистров, видимый непосредственно процессором.

     Модель SISC  процессора  является   модулем   "закрытой системы" без каких-либо портов входа или выхода и имеет следующий вид:

      module system;

                ...//  Элементы модели, включая описания,

                ...//  задачи, функции, блоки initial и

                ...//  always и т.п.

      endmodule    //  система

     В следующих разделах мы опишем каждый из элементов  модуля.

     Описания.

     Так как моделирование осуществляется на высоком уровне, мы должны оперировать в терминах регистров и разрядовых  полей  регистров,  а  не в терминах вентилей и переключателей. Некоторыми необходимыми нам регистрами являются:

      32-разрядный регистр для хранения команд

      12-разрядный регистр для адресации памяти

      5-разрядный регистр для признаков кодов условий

      33-разрядный регистр для хранения результатов.

     Эти и другие регистры, а также параметры описаны, как показано на рис. 3.3.

     Память и  регистровый  файл описаны как массивы регистров, размер которых равен WIDTH. Это описание позволяет осуществлять  произвольный  доступ ко всем ячейкам памяти и регистрового файла без моделирования протокола взаимодействия. Для  доступа  к ячейкам памяти или регистрового файла теперь можно просто указать структуру,  где индексом является соответствующая ячейка. Например,

     RFILE[3]=MEM[20];

передаст содержимое ячейки памяти 21 в 4-ую ячейку регистрового файла. Обратим внимание, что мы использовали 0 в качестве начального индекса в обеих структурах.

     Максимальный размер адресуемой памяти определяется наибольшим размером поля адреса (12 разрядов). Размер регистрового файла был произвольно выбран равным 16.

,---------------------------------------------.

|  //Описание  параметров                     |

|                                             |

| parameter    WIDTH  = 32 ;                  |

| parameter    CYCLE = 10 ;                   |

| parameter    ADDRSIZE = 12  ;               |

| parameter    MAXREGS   =   16  ;            |

| parameter    MEMSIZE  =(1<<ADDRSIZE);       |

|                                             |

|  //Описание регистров                       |

|                                             |

| reg [WIDTH-1:0] MEM[0:MEMSIZE-1],           |

|                 RFILE[0:MAXREGS-1],         |

|                 ir,  //                     |

|                 src1,  src2;                |

|reg[WIDTH:0]     result ;                    |

|reg [ADDRSIZE-1:0] pc ;                      |

|reg[4:0]         psr ;                       |

|reg              dir;                        |

|reg              reset;                      |

|                                             |

|integer          i;                          |

`---------------------------------------------'

Рис. 3.3 Описания для модели системы команд

     Регистр результата  выбран  33-разрядным  для  хранения разряда  переноса  после выполнения арифметических операций. Счетчик команд  содержит  адрес команды в памяти,  и поэтому его размер равен 12 разрядам.  Регистр состояния  процессора psr содержит 5 признаков кода условий: перенос (carry), четный результат  (even),  четность  результата  (parity),  равенство  нулю  (zero)  и  признак  отрицательного результата (negative).

,--------------------------------------------------.

| //Описание полей команды                         |

|                                                  |

| `define OPCODE    ir[31:28]                      |

| `define SRC       ir[23:12]                      |

| `define DST       ir[11:0]                       |

| `define SRCTYPE   ir[27]                         |

| `define DSTTYPE   ir[26]                         |

| `define CCODE     ir[27:24]                      |

| `define SRCNT     ir[23:12]                      |

|                                                  |

| // Типы операндов                                |

|                                                  |

| `define REGTYPE 0                                |

| `define IMMTYPE 1                                |

|                                                  |

| //  Определение кодов операций для каждой команды|

|                                                  |

| `define NOP 4'b0000                              |

| `define BRA 4'b0001                              |

| `define LD  4'b0010                              |

| `define STR 4'b0011                              |

| `define ADD 4'b0100                              |

| `define MUL 4'b0101                              |

| `define CMP 4'b0110                              |

| `define SHF 4'b0111                              |

| `define ROT 4'b1000                              |

| `define HLT 4'b1001                              |

`--------------------------------------------------'

Рис. 3.4  Определение символических имен для полей разрядов

     Оператор параметр   (parameter)   объявляет  символьные константы WIDTH и CYCLE.  Эти символьные константы дают возможность создавать легко адаптируемые модели. Например, если объем памяти равен 16 Кб,  то мы  должны  изменить  параметр ADDRSIZE с 12 на 14.  Это изменит значение параметра MEMSIZE с 4К до 16К,  и индекс массива памяти будет изменяться от  0 до 16К-1.

     Оператор 'define (определить),  показанный на рис. 3.4, позволяет ссылаться на различные поля регистра команд и  кодов условий с помощью символьных имен, а не чисел. Использование символьных имен для различных полей обеспечивает независимость  модели  от  размещения (расположения друг относительно друга) этих полей в регистре команд.  Например,  если поля SRCTYPE и DSTTYPE поменяются местами, то во всей модели необходимо будет изменить только 2 строки, а именно:

     `define SCRTYPE ir[26]

     `define DSTTYPE ir[27]

Рис. 3.4  демонстрирует  описание  символьных имен для полей разрядов.

     Основной процесс

     Для того,  чтобы  понять  процесс моделирования,  можно представить  его  себе  в  виде  трехступенчатого  процесса. Во-первых, должно быть выполнено определение основных объектов или структур.  Далее надо описать способ обработки  этих структур.  И  наконец,  убедиться,  что модель соответствует спецификации.

     В предыдущем разделе мы определили объекты, а именно: регистры, регистровый файл и память. Сейчас мы опишем способ их обработки.

     Без углубления в подробности выполнения  операций  процессором достаточно сказать, что процессор непрерывно выполняет  цикл  "выборка-исполнение".  Показанное  на  рис.  3.5 описание  на  Verilog  моделирует  основной  (main) процесс. Основной  процесс,  описанный  блоком  always,  помечен  как main_process и состоит из трех задач:  выборка (fetch),  выполнение (execution) и запись результата (write_result). Задача  fetch извлекает команды из памяти,  задача execute выполняет их,  и задача write_result  записывает  результат  в регистровый файл.  Эта  последовательность  повторяется  для всех команд.

,---------------------------------------------.

|     always begin : main_process             |

|       if (!reset) begin                     |

|           #CYCLE fetch;                     |

|           #CYCLE execute;                   |

|           #CYCLE write_result;              |

|       end                                   |

|       else  #CYCLE;                         |

|     end                                     |

`---------------------------------------------'

Рис. 3.5 Основной процесс

     Данная модель предполагает,  что эти задачи выполняются последовательно,  а  конвейерная  обработка и параллелизм не используются.  В  результате выполнение каждой команды занимает 3 такта.  В дальнейшем мы рассмотрим реализацию  конвейерной архитектуры.

     Сигнал сброса (reset) (точнее, регистр сброса) проверяется для определения того,  находится ли процессор в состоянии сброса.  Если да, то процесс main ждет один такт до следующей  проверки  сигнала  reset.  В  операторе if-then-else предложение else  является  необходимым.  Без  него  процесс always  войдет  в  бесконечный  цикл с нулевой задержкой при установленном сигнале reset.

ИНИЦИАЛИЗАЦИЯ СИСТЕМЫ

     Начальное состояние  при   моделировании   эквивалентно состоянию, возникающему при включении аппаратуры. Для инициализации системы загружается тестовая программа,  включается монитор  и  в  блоке инициализации используется установочная последовательность, как показано на рис. 3.6. Блок инициализации выполняется только один раз при начале моделирования.

,-------------------------------------------------.

|task apply_reset;                                |

|  begin                                          |

|    reset = 1 ;                                  |

|    #СYCLE ;                                     |

|    reset = 0 ;                                  |

|    pc = 0 ;                                     |

|  end                                            |

|endtask                                          |

|                                                 |

| initial begin  :  prog_load                     |

|         $readmemb("sisc.prog",MEM);             |

|         $monitor("%d %d %h %h %h",$time,pc,     |

|                RFILE[0],RFILE[1],RFILE[2]);     |

|         apply_reset;                            |

|  end                                            |

`-------------------------------------------------'

Рис. 3.6 Процесс инициализации

     Системная задача  Verilog  $readmemb  используется  для загрузки тестовой программы из файла данных ASCII  в  массив памяти mem.

     Установочная последовательность  включается  с  помощью задачи apply_reset (рис. 3.6). В данном примере установочная последовательность является очень простой,  она  переключает сигнал сброса и устанавливает счетчик команд в 0.

     Задача $monitor обеспечивает отслеживание информации  о времени  моделирования,  счетчике  команд и определенных регистров из регистрового файла.

     Функции и задачи

     После описаний,  процесса main и инициализации  системы следующий  шаг  состоит в том,  чтобы создать задачи и функции,обеспечивающие функционирование процессора.  Функциями в нашем описании являются getsrc, getdst и checkcond.  Задачами  являются  fetch,  execute,  write_results, set_condcode, clear_condcode и apply_reset.

     Задача execute  представляет  наибольший  интерес  (см.рис.3.7).  Она использует оператор выбора case для декодирования кода операции OPCODE и обеспечивает выполнение определенных действий,  которые соответствуют каждой отдельной команде.

     Обратите внимание на использование символьных имен  для доступа к различным полям регистра команд. Если расположение или длина полей должна быть изменена,  то  необходимо  будет скорректировать только описания,  при этом сами задачи останутся неизменными.

     Пустая операция (nop) увеличивает  время  моделирования на  один  такт без каких-либо других действий.  Команда halt выполняется с помощью системной задачи $stop.  Действия  при несовпадении  обеспечивает механизм обнаружения неправильных команд.  Команда branch показывает, как вызывается функция в Verilog.

     Команды load и store осуществляют доступ  к  памяти  по простому  протоколу.  Команды shift и rotate также реализуют довольно сложные аппаратные операции с помощью простых функциональных конструкций Verilog.

     Команды add и multiply на первый взгляд  кажутся  очень простыми.  Однако это не так.  Результат сложения двух чисел может быть на один разряд больше размера  большего  из  двух слагаемых, а размер произведения может быть равен сумме размеров множителя и множимого.  Назначение  произведению  двух чисел,  размер которых равен WIDTH, (WIDTH+1)-разрядного регистра сохраняет младшие WIDTH+1 разряды результата, и только  WIDTH младших разрядов сохраняются в памяти или в ячейке регистрового файла.

     Одно из  решений  этой  проблемы  состоит в том,  чтобы использовать для вычислений более длинные регистры (размером 2*WIDTH) и пару регистров или две ячейки памяти для хранения результата.  Другое решение - уменьшить размер операндов таким образом,  чтобы он не превышал половину размера поля результата.  Еще один подход - определить,  что точностью  результата являются младшие WIDTH+1 разряда.  В примере с SISC процессором мы выбрали точность результата в 32 разряда.

     Важно помнить,  что  в  Verilog арифметические операции выполняются в дополнительном коде.

,---------------------------------------------------------.

| task execute ;                                          |

| begin                                                   |

|   case (`OPCODE)                                        |

|     `NOP :;                                             |

|     `BRA : begin                                        |

|              if (checkcond(`CCODE)) pc = `DST ;         |

|            end                                          |

|     `HLT : begin                                        |

|              $display("Halt ..."); $stop;               |

|            end                                          |

|     `LD :  begin                                        |

|              clearcondcode ;                            |

|              if (`SRC) RFILE[`DST] =`SRC ;              |

|              else   RFILE[`DST]  =  MEM[`SRC];          |

|              setcondcode({1'b0,RFILE[`DST]}) ;          |

|            end                                          |

|     `STR : begin                                        |

|              clearcondcode ;                            |

|              if (`SRC) MEM[`DST] = `SRC ;               |

|              else MEM[`DST] = RFILE[`SRC] ;             |

|            end                                          |

|     `ADD : begin                                        |

|              clearcondcode ;                            |

|              src1 =getsrc(ir) ;                         |

|              src2 = getdst(ir) ;                        |

|              result =  src1  +  src2 ;                  |

|              setcondcode(result);                       |

|            end                                          |

|                 . . .                                   |

|                 . . .                                   |

|     default: $display("Invalid Opcode found");          |

|   endcase                                               |

|end                                                      |

|endtask                                                  |

`---------------------------------------------------------'

Рис. 3.7 Задача execute

     Тестовая программа

     Когда модель  написана,  мы должны подготовить тестовую программу для моделирования. Для этого используется программа,  подсчитывающая количество "1" в двоичном коде. Двоичный код такой программы представлен на рис.  3.21. Эта программа

хранится в файле sisc.prog и загружается в массив памяти mem с  помощью  системной  задачи  $readmemb.  Код  программы  в 16-ричном виде может быть загружен с помощью системной задачи $readmemh.

     Запуск модели

     С помощью тестовой программы для нашей модели мы  можем продемонстрировать,  каким образом команды выбираются из памяти,  дешифрируются и выполняются, а также как запоминаются результаты. Эта имитирующая модель сейчас может быть использована для разработки дополнительной диагностики, системного программного обеспечения или прикладных программ для целевой аппаратуры .  Начать нам позволит следующая команда в  ответ на приглашение операционной системы:

         %verilog sisc_instruction_set_model.v

где sisc_instruction_set_model.v является именем файла,  содержащего  описание  на  Verilog  модели системы команд SISC процессора.

     Существуют две  стадии  при   моделировании   в   среде Verilog. Первая состоит в устранении всех ошибок, выявляемых при  трансляции  и  установке  связей.   Ошибки   трансляции обусловлены в основном синтаксическими ошибками и неописанными структурами.  Ошибки связей связаны с размерами  порта, созданием модуля, потерей связей и т.п.

     Синтаксически правильное описание модели на Verilog теперь может быть промоделировано для определения правильности работы.  Обычно  этот  этап занимает гораздо больше времени, чем первый. Использование диалоговой отладки и некоторых уже

испробованных средств может значительно сократить это время. Некоторые полезные отладочные средства будут описаны ниже.

ОТЛАДКА

     Хотя данная  модель написана достаточно правильно,  без процесса отладки можно обойтись лишь при  простейших  разработках. Кроме диалогового отладчика, который предоставляется в распоряжение  пользователя  моделирующей  программой,  нам потребуются  некоторые обычные для тестирования модели задачи. Для данной модели важно проследить за изменением счетчика команд и определенных регистров в регистровом файле.

     Пример задачи disprm,  предназначенной  для  облегчения отладки модели,  показан на рис.  3.8.  Она используется для отображения содержимого  некоторого  диапазона  регистров  и ячеек  памяти.  Достаточно  неудобно  набирать в теле задачи соответствующую команду каждый раз, когда необходимо посмотреть содержимое ячеек памяти,  скажем, 20-ой, 21-ой и 22-ой; кроме того,  набор на клавиатуре отдельных  команд  $display также занимает много времени, если конечный адрес значительно отличается от начального.

,----------------------------------------------------.

|// Для  отображения содержимого регистров           |

|// и ячеек памяти                                   |

| task  disprm  ;                                    |

| input  rm  ;                                       |

| input  [ADDRSIZE-1:0]  adr1,  adr2  ;              |

| begin                                              |

|  if (rm ==`REGTYPE)  begin                         |

|      while  (adr2  >=  adr1)  begin                |

|             $display("REGFILE[%d]=%d\n",           |

|                       adr1,RFILE[adr1]) ;          |

|       adr1 = adr1 + 1 ;                            |

|      end                                           |

|  end                                               |

|  else begin                                        |

|      while  (adr2  >=  adr1)  begin                |

|            $display("MEM[%d]=%d\n",                |

|                      adr1,MEM[adr1]) ;             |

|          adr1 = adr1 + 1 ;                         |

|      end                                           |

|  end                                               |

|end                                                 |

|endtask                                             |

`----------------------------------------------------'

Рис. 3.8 Задача для печати состояний при отладке

     Задача apply_reset используется для перезапуска моделирования без  необходимости выхода и нового входа.  Она может быть значительно улучшена установкой всех регистров и  ячеек памяти в неопределенное состояние (х), ноль или состояние заранее определенного набора разрядов. Задача apply_reset также  может  быть  изменена  для  включения  системной  задачи $readmemb,  чтобы при каждом запуске системы загружалась новая программа.

МОДЕЛИРОВАНИЕ УПРАВЛЕНИЯ КОНВЕЙЕРОМ

     До этого  момента  мы  моделировали SISC архитектуру на уровне системы команд.  Сейчас  давайте  обсудим  управление процессором.  Мы кратко опишем принцип конвейерной обработки для создания функциональной модели путем внесения  изменений

в уже разработанную модель SISC архитектуры.

     Что такое конвейер ?

     В основном   конвейерная  архитектура  применяется  для использования внутреннего параллелизма или  для  обеспечения ресурсами  при необходимости введения параллелизма.  Простой конвейер может быть основан на одновременности выборки одной команды из памяти и выполнения выбранной перед этим команды.

     Другими словами, когда выполняется одна команда, следующая считывается из памяти. Если следующая команда уже готова к выполнению к моменту окончания текущей команды, то происходит  одновременное выполнение цикла выборки и выполнения команды. Это пример командного конвейера.

,----------------------------------------------------.

| такт  #  без конвейера     с конвейером            |

|  1           F1              F1                    |

|  2           E1              F2  E1                |

|  3           W1              F3  E2  W1            |

|  4           F2              F4  E3  W2            |

|  5           E2              F5  E4  W3            |

|  6           W2                  E5  W4            |

|  7           F3                      W5            |

|  8           E3                                    |

|  9           W3                                    |

|  10          F4                                    |

|  11          E4                                    |

|  12          W4              F12                   |

|  13          F5              F13 E12 W11           |

|  14          E5              F14 E13 W12           |

|  15          W5              F15 E14 W13           |

|                                                    |

| Без конвейера 5 команд выполняются за 15 тактов    |

| (кол-во команд)*3                                  |

| С конвейером 5 команд выполняются  за  7  тактов   |

| (кол-во команд)+2                                  |

|                                                    |

|Обозначения: F=выборка, Е=выполнение, W=запись      |

|             результата                             |

`----------------------------------------------------'

Рис. 3.9 Табличное представление 3-х уровневого  конвейера

     Рис. 3.9 поясняет, каким образом будет обрабатывать команды 3-х уровневая конвейерная архитектура. Уровнями  являются выборка (F),  выполнение (Е) и запись (W).  Предполагается,  что  для  выполнения  на каждом уровне необходим один такт. Таким образом, процессор без конвейерной обработки выполняет  каждую команду за три такта,  тогда как процессор с конвейером может выполнять в  среднем  одну  команду  каждый такт. Для увеличения производительности может использоваться конвейер с большим количеством уровней.

     Для повышения  производительности следующая команда готовится к выполнению (локальность ссылок).  Однако,  даже  в достаточно  простых  случаях  существуют  определенные сложности. Например, выполнение условий команды перехода означает, что за командой перехода должна выполняться не следующая по адресу,  а совсем другая команда. Поэтому та команда, которая  уже была выбрана (заранее выбранная команда), должна быть уничтожена,  и должна быть выбрана та команда, на которую происходит переход.  Так как процессор должен уничтожить заранее выбранную команду, исполнительное устройство простаивает  в  течении  одного или большего количества тактов ( в зависимости от того, сколько тактов занимает выборка команды из памяти).

     В процессоре с большим количеством  уровней   конвейера могут  потребоваться  более сложные действия для правильного завершения команд в конвейере.  Это явление обычно  называют "очисткой конвейера". Команда halt также очищает конвейер.

     Другой причиной снижения эффективности конвейера  являются  команды  load и store.  Так как эти команды необходимы для доступа в память и передачи данных в или из регистрового файла внутри процессора, процессор не может выбирать следующую команду.  Он пропускает цикл выборки, что в свою очередь вызывает  потерю  цикла  выполнения,  если у процессора есть исходная информация только для  заранее  выбранной  команды. Для заполнения конвейера используются многопортовые память и регистровый файл.

     Функциональное разбиение

     Сейчас давайте усовершенствуем разработанную ранее SISC модель, добавив в нее 3-х уровневый конвейер. Тремя уровнями конвейерной обработки являются выборка команды (fetch),  выполнение    команды    (execute)    и    запись   результата (write_result).  Когда выполняется текущая команда, происходит  выборка  следующей  команды и запись в регистровый файл результата предыдущей команды.  Но этот  порядок  нарушается при командах перехода (brаnch), считывания из памяти (load) и записи в память (store).

,------------------------------------------------.

| always @(posedge clock) begin : phase1_loop    |

|   if (!reset) begin                            |

|      fetched  =  0 ;                           |

|      executed  = 0 ;                           |

|   if (!queue_full && !mem_access)              |

|       -> do_fetch ;                            |

|   if (qsize || mem_access)                     |

|       ->  do_execute  ;                        |

|   if  (result_ready)                           |

|       ->  do_write_results  ;                  |

|   end                                          |

| end                                            |

`------------------------------------------------'

Рис. 3.10 Запуск одновременных событий

     Мы упростили конвейер SISC процессора, сделав его синхронным.  Как видно из рис.  3.10 по переднему  фронту  синхросигнала (по фазе 1,  если генератор тактовых импульсов является двухфазным) одновременно порождаются  три  события  - do_fetch, do_execute и do_write_result. Блок phase1_loop получается изменением блока main_process  разработанной  ранее SISC модели. По отрицательному фронту синхросигнала информация передается между уровнями конвейера.  На рис. 3.11 показаны дополнительные описания переменных, которые потребуются для модели конвейера. Последующее объяснение модели конвейера  будет  содержать  ссылки на эти регистры (reg),  провода(wire) и события (event).

,-------------------------------------------------------.

| parameter  QDEPTH=3; // Длина очереди команд          |

|  // Очередь команд и регистр команд для записи        |

| reg [WIDTH-1:0]     IR_Queue[0:QDEPTH-1],wir;         |

|                                                       |

|  // Копия результата и указатели выборки и выполнения |

| reg [WIDTH:0]  wresult;                               |

| reg [2:0]   eptr, fptr, qsize ;                       |

|                                                       |

|  // Различные управляющие сигналы/признаки            |

| reg     mem_access, branch_taken, halt_found ;        |

| reg     result_ready ;                                |

| reg     executed, fetched ;                           |

| reg     queue_full ;                                  |

|                                                       |

| event   do_fetch, do_execute, do_write_results;       |

`-------------------------------------------------------'

Рис. 3.11  Дополнительные  описания  для  моделирования конвейера

     Обратите внимание,  что  порядок  операторов if в блоке phase1_loop не является существенным,  так как эти операторы только порождают события, а соответствующие события и задачи не обязательно выполняются в этом порядке. Эти события моделируют работу трех функциональных модулей программы.

     Устройство выборки

     Основная функция устройства выборки состоит в том, что бы передать команду  из  памяти  в  процессор  и  обеспечить доступ исполнительного устройства к ней.  Чтобы исполнительное устройство не простаивало,  необходимо организовать очередь  команд с предварительной их выборкой.  Это достигается использованием очереди IR_Queue.  Мы выбрали  длину  очереди равную трем,  что соответствует количеству уровней  конвейера.  2-х разрядный регистр fptr содержит ссылку  на  текущую позицию  в очереди IR_Queue,  в которую должна быть помещена последняя выбранная команда. Содержимое регистра qsize определяет  количество  заранее  выбранных команд.  Этот регистр используется также для определения того, является ли очередь IR_Queue заполненной или пустой.  Если в очереди нет команд, уровень выполнения команд не может  работать,  что  вызывает пропуск циклов выполнения команд и задержку в  конвейере;  в то же время,  если очередь IR_Queue полностью заполнена,  то считывание из памяти новой команды будет затирать еще не выполненную команду в очереди.

     Признак mem_access  дает  сигнал  пропуска  такта  устройству выборки при выполнении команды load или  store.  Это отражает оператор if,  управляющий событием do_fetch. Задача fetch (см. рис. 3.12) была изменена так, чтобы выбранная команда  помещалась в очередь IR_Queue,  а не записывалась непосредственно в регистр команд IR. При записи команды в очередь также устанавливается в "1"  признак выборки,  указывая тем самым на завершение цикла выборки.

,-----------------------------------------.

| task fetch;                             |

| begin                                   |

|     IR_Queue[fptr]=MEM[pc];             |

|     fetched=1;                          |

| end                                     |

| endtask                                 |

`-----------------------------------------'

Рис. 3.12  Задача выборки fetch

     Исполнительное устройство

     Исполнительное устройство подобно описанному ранее. Оно дешифрирует и выполняет текущую команду. Предполагается, что все команды,  кроме load и store,  выполняются за один такт. Предположим также следующее:  все команды занимают в  памяти одно слово,  операнды всех арифметических команд - сложения, умножения, дополнения, сдвига и циклического сдвига - являются либо непосредственными (заданными в слове команды), либо хранятся в регистровом файле.

     При дальнейшем  изучении арифметических операций оказывается,  что в АЛУ необходимы два входных регистра - src1  и src2 и один регистр на выходе - result.  Так как адреса операндов и адрес результата арифметических команд являются адресами в регистровом файле, последний должен иметь три независимых порта:  два для считывания операндов и один для  записи результата.  Напомним, что результат предыдущей команды записывается в регистровый  файл одновременно со считыванием операндов для текущей команды.

     В рассматриваемой архитектуре до сих пор не учитывалась память.  Работа  с памятью имитируется в исполнительном устройстве командами load и store, занимающими по два такта. На рис.  3.13 представлена часть этой модели, показывающая, как моделируется  команда  load.  При  выполнении  команды  load исполнительное   устройство   устанавливает  в  "1"  признак mem_access для резервирования доступа в память  в  следующем такте.  Устройство выборки команды в текущем такте завершает выборку команды из памяти.  В следующем такте оно простаивает, а исполнительное устройство обратится в память. Исполнительное устройство выполняет команду load предыдущего такта, так  как  в регистр команд не записывается новая информация, пока признак mem_access находится в состоянии "1".  По окончании  обмена с памятью исполнительное устройство сбрасывает признак mem_access,  позволяя тем самым себе записать в  регистр команд следующую команду из очереди,  а устройству выборки пополнить очередь новыми командами из памяти.

     Исполнительное устройство пропускает такт при  выполнении перехода или,  если очередь команд пуста. Очередь команд в SISC процессоре становится пустой только в  двух  случаях: при  переходе  (branch)  и в случае команды останова (halt). Показанная на рис.  3.7 задача flush_queue описывает необходимые при этом действия.

,-------------------------------------------------------.

|if (!mem_access) ir=IR_Queue[eptr];                    |

| `LD     : begin                                       |

|           if (mem_access == 0)    // Резервирование   |

|               mem_access = 1 ;    // следующего такта |

|           else begin              // Доступ к памяти  |

|                 ......            // в следующем      |

|                                   // такте            |

|           end                                         |

|end                                                    |

`-------------------------------------------------------'

Рис. 3.13 Доступ к памяти в случае команды load

,-------------------------------------------------.

| task flush_queue ;                              |

| begin                                           |

|   // pc уже изменен выполнением команды branch  |

|   fptr = 0 ;                                    |

|   eptr = 0 ;                                    |

|   qsize = 0 ;                                   |

|   branch_taken = 0 ;                            |

| end                                             |

| endtask                                         |

`-------------------------------------------------'

Рис. 3.14 Очистка конвейера

     Остальная часть модели исполнительного устройства остается неизменной для всех арифметических операций,  как показано  в  задаче  task  на  рис.  3.7.  Задачи  set_condcode, clear_condcode  и  функция checkcond также не изменяются при добавлении конвейерной обработки.

     Устройство записи результатов

     Уровни  выполнения и записи результата  связаны  с  помощью  двух регистров:  wresult и wir.  Если результат нужно записать в регистровый файл, исполнительное устройство устанавливает признак result_ready.  Результат переписывается из регистра результата result в регистр wresult,  а  содержимое регистра  команд ir - в wir.  Устройство записи переписывает результат из регистра wresult по адресу назначения в регистровом  файле,  который  определяется в следующем такте полем dest в регистре wir.  Так как wir является  копией  команды, при  выполнении  которой был получен данный результат,  поле dest содержит адрес назначения в регистровом файле для  пре-

дыдущей команды.

     Следует отметить,  что мы упростили устройство записи с помощью  копирования  текущей  команды  из регистра ir в wir вместо того,  чтобы организовывать указатель в  очереди  команд.  Задача copy_result (см.  рис. 3.15) определяет, каким образом результат из исполнительного устройства передается в устройство записи,  а рис.  3.16 показывает,  как происходит запись результата по адресу назначения в регистровом файле.

,----------------------------------------------------.

| task copy_results ;                                |

| begin                                              |

|   if ((`OPCODE >= `ADD) && (`OPCODE < `HLT)) begin |

|      setcondcode(result) ;                         |

|      wresult = result ;                            |

|      wir = ir ;                                    |

|      result_ready = 1 ;                            |

|   end                                              |

| end                                                |

| endtask                                            |

`----------------------------------------------------'

Рис. 3.15 Копирование результата и команды

     Существует два альтернативных подхода,  исключающие необходимость копирования результата из  одного  регистра  результата в другой.  Один подход состоит в том, чтобы результат записывался в регистр результата с помощью  исполнительного устройства во время отрицательной фазы синхросигнала, а содержимое регистра результата  копировалось  в  регистровый файл  во время положительной фазы следующего такта с помощью устройства записи.  Еще одна возможность - полностью удалить уровень записи в конвейере и записывать результат из АЛУ непосредственно в регистровый файл.  С  помощью  моделирования можно исследовать каждый из этих подходов.

,--------------------------------------------------------.

| task write_result ;                                    |

| begin                                                  |

|   if ((`WOPCODE >= `ADD) && (`WOPCODE < `HLT)) begin   |

|      if (`WDSTTYPE == `REGTYPE)                        |

|           RFILE[`WDST] = wresult;                      |

|      else MEM[`WDST] = wresult ;                       |

|      result_ready = 0 ;                                |

|   end                                                  |

| end                                                    |

| endtask                                                |

`--------------------------------------------------------'

Рис. 3.16 Запись результата в регистровый файл

     Операции управления фазы 2

     В общем  SISC  процессор  является  синхронной  схемой. Описанные выше функциональные блоки соответствуют трем уровням конвейерной обработки. Во время положительной фазы синхроимпульса происходит следующее:

     * изменяется счетчик команд (рс) в зависимости от того,

       выполняется или нет команда перехода.

     * устанавливаются коды условий вновь вычисленного

       результата.

     * копируется содержимое регистра команд (ir) и регистра

       результата в теневые регистры (wir и wresult  соот-

       ветственно).

     * обновляются  указатели выборки  и  выполнения (eptr и

       fptr) в очереди команд.

,--------------------------------------------------------.

| task set_pointers ;  // Управление указателями очереди |

| begin                                                  |

|    case ({fetched,executed})                           |

|     2'b00 : ;            // Пропуск такта выборки      |

|     2'b01 : begin       // Нет выборки                 |

|                qsize = qsize - 1 ;                     |

|                eptr = (eptr + 1)%QDEPTH ;              |

|             end                                        |

|     2'b10 : begin       // Нет выполнения              |

|                qsize = qsize + 1 ;                     |

|                fptr = (fptr + 1) % QDEPTH ;            |

|             end                                        |

|     2'b11 : begin       // Выборка и выполнение        |

|                 eptr = (eptr + 1)%QDEPTH ;             |

|                 fptr = (fptr + 1) % QDEPTH ;           |

|             end                                        |

|    endcase                                             |

| end                                                    |

| endtask                                                |

| always @(negedge clock) begin : phase2_block           |

|   if (!reset) begin                                    |

|      if (!mem_access && !branch_taken)                 |

|         copy_results ;                                 |

|      if (branch_taken) pc = `DST ;                     |

|      else if (!mem_access) pc = pc+1;...               |

|      if (branch_taken || halt_found)                   |

|         flush_queue ;                                  |

|      else set_pointers ;                               |

|      if (halt_found) begin                             |

|          $stop ;                                       |

|          halt_found = 0 ;                              |

|      end                                               |

|   end                                                  |

| end                                                    |

`--------------------------------------------------------'

Рис. 3.17 Операции управления фазы 2

     Все эти действия показаны на рис. 3.17. Они выполняются задачей set_pointers и процессом phase2_loop.

     Очередь команд необходимо очищать, но эта процедура откладывается  до  отрицательной половины такта,  так как надо закончить выполнение команды,  находящейся в  конвейере.  По той  же причине в случае команды halt моделирование не останавливается до тех пор, пока не будут проделаны все действия в  отрицательной  половине такта.  Это дает возможность устройству записи запоминать результат  предыдущей  команды  во время положительной половины следующего такта,  когда исполнительное  устройство  устанавливает  одноразрядный  признак halt_found.

     Проблема взаимных блокировок

     В модели конвейерной архитектуры существует неочевидная проблема.  Ее обычно называют  "блокировкой  регистра".  Эта проблема проявляется, когда за командой, изменяющей содержимое какого-либо регистра в регистровом файле, следует команда, которая пытается произвести считывание из того же самого регистра.  В конвейерной архитектуре с  большим  количеством уровней  проявление  этого факта  может быть более сложным и трудно распознаваемым.

     Ситуацию блокировки  регистра лучше объяснить на примере.  Рассмотрим два программных сегмента, показанные на рис. 3.18.  В  первом  сегменте  не  существует  конфликта  из-за ресурса - регистра R2 в регистровом файле при  условии,  что регистровый файл является многопортовым.

     Ситуация изменяется  с  вторым  программным  сегментом. Первая команда (I3) считывает содержимое двух регистров  (R1 и  R2) из регистрового файла и записывает результат сложения в регистр R1.  Вторая команда (I4) считывает содержимое  регистра  R1 и запоминает его дополнение  в регистре R3. Из-за одновременной работы исполнительного устройства и устройства записи исполнительное устройство может считать неверное значение из регистра R1 во время выполнения  команды  I4.  (Результат  команды I3 в это время будет записываться в регистр R1.)

,-------------------------------------------.

|         Первый программный сегмент        |

|    I1:    ADD    R1, R2   // R1= R1 + R2  |

|    I2:    CMP    R3, R2   // R3= ~R2      |

|                                           |

|         Второй программный сегмент        |

|    I3:    ADD    R1, R2   // R1= R1 + R2  |

|    I4:    CMP    R3, R1   // R3 = ~R1     |

`-------------------------------------------'

Рис. 3.18 Проблема блокировки

     Если величина из R1 считывается до того, как закончится запись результата,  I4 будет оперировать с устаревшим содержимым регистра R1.  В реальной аппаратуре это может привести к сбою или к гонке сигналов в зависимости от реализации  команд  записи и чтения для многопортового регистрового файла. Ниже будут рассмотрены два возможных решения.

     Простейшее решение  - вставить пустую команду nop между двумя командами, вызывающими блокировку регистра. Преимущество этого решения состоит в том,  что не нужно изменять архитектуру и проект.  Однако, есть и недостаток: выполнение дополнительных команд,  вызывающее потерю одного такта на каждую блокировку регистра и уменьшение производительности. Это также  потребует  изменений в математическом обеспечении,  в частности,  в SISC компиляторе и оптимизаторе. Модифицированный второй сегмент программы представлен на рис. 3.19.

,-----------------------------------------------------.

|    I3:    ADD    R1, R2     // R1 = R1 + R2         |

|    IX:    NOP               // Избежание блокировки |

|    I4:    CMP    R3, R1     // R3= ~R1              |

`-----------------------------------------------------'

Рис. 3.19 Измененный второй программный сегмент

,--------------------------------------------------.

| reg    bypass;                                   |

|                                                  |

| function [31:0] getsrc;                          |

| input [31:0] i ;                                 |

| begin                                            |

|   if (bypass) getsrc = result ;                  |

|   else if (`SRCTYPE === `REGTYPE)                |

|       getsrc = RFILE[`SRC] ;                     |

|   else getsrc = `SRC ; // непосредственный тип   |

| end                                              |

| endfunction                                      |

|                                                  |

| function [31:0] getdst;                          |

| input [31:0] i ;                                 |

| begin                                            |

|   if (bypass) getdst=result;                     |

|   else if  (`DSTTYPE === `REGTYPE)               |

|        getdst = RFILE[`DST] ;                    |

|   else $display("Error : Immediate data          |

|                  cannot be destination.");       |

| end                                              |

| endfunction                                      |

|                                                  |

| always @(do_execute) begin : execute_block       |

|      if (!mem_access ) begin                     |

|        ir=IR_Queue[eptr];                        |

|        bypass=((`SCR == `WDST) ||                |

|                (`DST == `WDST));                 |

|      end                                         |

|      execute ;                                   |

|      if (!mem_access) executed = 1;              |

| end                                              |

`--------------------------------------------------'

Рис. 3.20 Измененные функции и процесс выполнения для метода обхода

     Другой метод состоит в реализации в схеме  дополнительной логики для распознавания условий регистровой блокировки. После определения ситуации аппаратура может  копировать  содержимое  регистра  результата  предыдущей команды в один из операндов (src1 или src2) текущей команды.  В  то  же  время должно быть предотвращено чтение операнда из заблокированного регистра для избежания  затирания  правильного  значения. Этот метод известен как метод "обхода".  Часть модели, показанная на рис.  3.20, реализует метод обхода в SISC архитектуре для предотвращения блокировки.

     Следует рассмотреть случай,  подобный этому,  когда регистр назначения команды используется как источник  для  команды store,  следующей за данной командой. Анализ результатов моделирования следует использовать для определения того, какой из рассмотренных подходов лучше.

ГЕНЕРАЦИЯ ТЕСТОВЫХ ВЕКТОРОВ

     После того,  как  написана  модель  какой-либо системы, важно убедиться в  том,  что  она  написана  правильно.  При тестировании выявляются некоторые неочевидные ситуации, связанные с архитектурой или реализацией,  например регистровая блокировка. Это одинаково верно для моделей от простого вентиля NAND до сложного микропроцессора и до значительно более сложных  микропроцессорных систем и даже сетей таких систем. Если модели небольшие и  легки  для  понимания,  применяются простые  методы  проверки правильности написания моделей.  В этом случае возможно полное тестирование.  В случае  сложных систем  не всегда можно сгенерировать все возможные комбинации.  В качестве тестовых векторов  для  проверки  отдельных частей  системы  используется  большое количество диагностических программ.

     Эти диагностические  программы могут  быть  написаны на языке программирования высокого уровня,  таком как Си, и написанный  для  конкретной системы транслятор может генерировать соответствующий объектный (машинный) код. Если в распоряжении нет транслятора,  то для создания машинного кода закодированных вручную программ может быть использован простой ассемблер для конечной системы команд. Такие средства облегчают выполнение промежуточной задачи создания тестовых  векторов  из "0" и "1",  при выполнении которой очень легко допустить ошибку и которая отнимает очень много  времени.  Еще один  недостаток  ручного  кодирования часто проявляется при изменении формата команд (порядка различных полей  в  команде).

     После трансляции диагностической программы в последовательность  команд эта программа загружается в память,  отведенную для моделирования,  точно  так  же,  как  выполняемая программа загружается в системную память. Запуск системы вызывает извлечение команды из памяти,  начиная с команды,  на которую  указывает счетчик команд.  Сама по себе команда подобна подаче тестового вектора на выводы процессора.  Выполнение  команды  в процессоре управляет извлечением следующих команд (тестовых векторов).

     Моделирующая программа Verilog-XL может загружать диагностическую программу, записанную в ASCII файле, в моделируемую  память  с  помощью  одной  из  двух  системных  задач: $readmemb или $readmemh. Задача $readmemb используется, когда команды представлены в двоичном виде, а $readmemh – когда в 16-ричном.Для удобства чтения файлы программы могут содержать комментарии, символы подчеркивания (_) и пробелы.

     Обычно программа загружается в  моделируемую  память  в начале  моделирования.  Это  происходит при помощи системной задачи $readmemb в операторе initial (см.  на рис. 3.6 блок, помеченный  как prog_load).  Образец программы,  которая вычисляет количество "1" в двоичном коде - на рис. 3.21. Здесь конец программы отмечен остановкой моделирования для возможной проверки содержимого регистров,  ячеек памяти  и  других сигналов и переменных.  Для ускорения процесса верификации и выполнения  программы  без  ошибок  существует  общепринятая практика  сравнения вычисленных результатов с ожидаемыми.  В диагностической программе,  содержащей сотни или тысячи  команд,  такие  сравнения  могут  быть разбросаны по различным частям программы.

----------------------------------------------------------.

//   Программа для подсчета количества '1                 |

//   в заданном двоичном числе                            |

//                                                        |

0010_1000_0000_0000_0000_0000_0000_0001//     LD  R1,#0   |

0010_0000_0000_0000_1001_0000_0000_0000//     LD  R2,NMBR |

0001_0010_0000_0000_0000_0000_0000_0100//STRT:BRA L1      |

0100_1000_0000_0000_0001_0000_0000_0001//     ADD R1,#1   |

0111_1000_0000_0000_0001_0000_0000_0000//L1  :SHF R2,#1   |

0001_0100_0000_0000_0000_0000_0000_0111//     BRA L2,ZERO |

0001_0000_0000_0000_0000_0000_0000_0010//     BRA STRT,ALW|

0011_0000_0000_0000_0001_0000_0000_1010//L2  :STR RSLT, R2|

1001_1111_1111_1111_1111_1111_1111_1111//     HLT         |

0101_0101_0101_0101_1010_1010_1010_1010//NMBR:5555аааa    |

0000_0000_0000_0000_0000_0000_0000_0000//RSLT:00000000    |

----------------------------------------------------------'

Рис. 3.21 Программа на ассемблере в качестве тестового вектора

     Когда имитационная модель является очень большой, выход и повторный вход в моделирующую программу для загрузки новой диагностической программы может занимать много времени. Если смоделировать поведение сигнала reset для возвращения машины в первоначальное состояние, системная задача $readmemb может быть  использована  в  диалоговом  режиме для загрузки новой программы при окончании работы текущей программы.

РЕЗЮМЕ

     В этой главе было показано,  как моделировать процессор на СБИС с помощью ЯОА Verilog. Модель была усовершенствована от первоначального описания  архитектуры  без  конвейера  до функционального  описания на высоком уровне с тремя  уровнями конвейерной обработки. Была подробно рассмотрена проблема регистровой блокировки для того, чтобы показать пользу моделирования при решении проблем или  принятии  конструкторских решений.  Были также представлены понятия создания тестового вектора.

     Полные описания обеих этих моделей на ЯОА Verilog -  на рис. 3.22. Первая часть содержит описания, функции и задачи, общие для двух моделей.  Следующая часть является  описанием модели без конвейера,  за которым следует модель процессора, включающая конвейерную обработку.

/*

 *  Описания, функции и задачи

 *  общие для обеих SISC моделей.

 *  sisc_declarations.v

 *

 *  %W% %G%  -- для управления версиями

*/

// Описания параметров

parameter       CYCLE = 10 ;            // Время такта

parameter       WIDTH = 32 ;            //Размер шины данных

parameter       ADDRSIZE = 12 ;         // Размер полей адреса
parameter       MEMSIZE = (1<<ADDRSIZE);//Максимальный
                                        //размер памяти

parameter       MAXREGS = 16 ; //Максимальное количество

                               //регистров в регистровом

                               //файле

parameter       SBITS = 5 ;    //Разряды регистра состояния

// Описание регистров и памяти

reg [WIDTH-1:0] MEM[0:MEMSIZE-1],   // Память
                RFILE[0:MAXREGS-1], // Регистровый файл

                ir,                 //Регистр команд

                src1, src2 ;        //Регистры операндов АЛУ

reg [WIDTH:0]   result ;            //Регистр результата АЛУ

reg [SBITS-1:0] psr ;       //Регистр состояния процессора

reg [ADDRSIZE-1:0]  pc ;    //Счетчик команд

reg             dir ;       //направление циклического сдвига

reg             reset ;     //Сброс системы

integer     i ;            // для диалоговой отладки

// Основные определения

`define TRUE   1

`define FALSE  0

`define DEBUG_ON       debug=1

`define DEBUG_OFF      debug=0

//Определения полей команды

`define OPCODE  ir[31:28]

`define SRC     ir[23:12]

`define DST     ir[11:0]

`define SRCTYPE ir[27] //тип источника,0=регистр (память

                       // для load),1=непосредственный

                       //операнд (imm)

`define DSTTYPE ir[26] //тип назначения, 0=регистр, 1=imm

`define CCODE   ir[27:24]

`define SRCNT   ir[23:12] //Величина сдвига для команд shift/

                          //rotate -=влево, +=вправо

// Типы операндов
`define REGTYPE 0

`define IMMTYPE 1

//Определения кодов операций для каждой команды

`define NOP   4'b0000

`define BRA   4'b0001

`define LD    4'b0010

`define STR   4'b0011

`define ADD   4'b0100

`define MUL   4'b0101

`define CMP   4'b0110

`define SHF   4'b0111

`define ROT   4'b1000

`define HLT   4'b1001

//Определение полей кодов условий

`define CARRY   psr[0]

`define EVEN    psr[1]

`define PARITY  psr[2]

`define ZERO    psr[3]

`define NEG     psr[4]

//Определение кодов условий

                //Код условия устанавливается, когда ...

`define CCС 0   //При вычислении результата был перенос

`define CCE 1   //Результат - четное число

`define CCP 2   //Cвертка результата по четности равна

                //  единице

`define CCZ 3   //Результат равен нулю

`define CCN 4   //Результат меньше нуля

`define CCA 5   //Всегда
`define RIGHT   0   // Сдвиг вправо в случае команд

                    // Rotate/Shift

`define LEFT    1   // Сдвиг влево в случае команд
                    // Rotate/Shift

// Функции для операндов и результата АЛУ

function [WIDTH-1:0] getsrc;

input [WIDTH-1:0] in ;

begin

  if (`SRCTYPE === `REGTYPE) begin

     getsrc = RFILE[`SRC] ;

  end

  else begin // непосредственный тип

      getsrc = `SRC ;

  end

end

endfunction

function [WIDTH-1:0] getdst;

input [WIDTH-1:0] in ;

begin

  if (`DSTTYPE === `REGTYPE) begin

      getdst = RFILE[`DST] ;

  end

  else begin //непосредственный тип

    $display("Error:Immediate data cant be destination.") ;

  end

end

endfunction

// Функции/задачи для кодов условий

function checkcond; // выдает 1, если код условия установлен

input [4:0] ccode ;

begin

  case (ccode)

   `CCA : checkcond = 1 ;

   `CCC : checkcond = `CARRY ;

   `CCE : checkcond = `EVEN ;

   `CCP : checkcond = `PARITY ;

   `CCZ : checkcond = `ZERO ;

   `CCN : checkcond = `NEG ;

  endcase

end

endfunction

task clearcondcode ;  // Сбрасывает коды условий в PSR

begin

  psr = 0 ;

end

endtask

task setcondcode ;  // Вычисляет и устанавливает

                    // коды условий
input [WIDTH:0] res ;

begin

  `CARRY = res[WIDTH] ;

  `EVEN  = ~res[0] ;

  `PARITY = ^res ;

  `ZERO  = ~(|res) ;

  `NEG = res[WIDTH-1] ;

end

endtask

/*=================================================

 *

 * SISC модель без конвейера

 *

 * sisc_instruction_set_model.v

 * Полезна для моделирования системы команд

 * Три главных задачи - fetch,execute,write.

 *

 * %W% %G% -- Для управления версией

*/

module instruction_set_model;

// Здесь необходимо включить файл sisc_declarations.v

// В этом модуле справедливы все общие описания,

// работают все общие функции и задачи.

// Остальная часть модели представлена ниже.

// Главные задачи - fetch, execute, write_result

task fetch ; //Выбрать команду и увеличить на 1

             // счетчик команд

begin

  ir = MEM[pc] ;

  pc = pc + 1 ;

end

endtask

task execute ;// Дешифрировать и выполнить команду

begin

  case (`OPCODE)

    `NOP    : ;

    `BRA    : begin

               if (checkcond(`CCODE) == 1) pc = `DST ;

              end

    `LD     : begin

               clearcondcode ;

               if (`SRC) RFILE[`DST] = `SRC ;

               else RFILE[`DST] = MEM[`SRC] ;

               setcondcode({1'b0,RFILE[`DST]}) ;

              end

    `STR    : begin

               clearcondcode ;

               if (`SRC) MEM[`DST] = `SRC ;

               else MEM[`DST] = RFILE[`SRC] ;

              end

    `ADD    : begin

               clearcondcode ;

               src1 = getsrc(ir) ;

               src2 = getdst(ir) ;

               result = src1 + src2 ;

               setcondcode(result) ;

              end

    `MUL    : begin

               clearcondcode ;

               src1 = getsrc(ir) ;

               src2 = getdst(ir) ;

               result = src1 * src2 ;

               setcondcode(result) ;

              end

    `CMP    : begin

               clearcondcode ;

               src1 = getsrc(ir) ;

               result = ~src1 ;

               setcondcode(result) ;

              end

    `SHF    : begin

               clearcondcode ;

               src1 = getsrc(ir) ;

               src2 = getdst(ir) ;

               i = src1[ADDRSIZE-1:0] ;

               result = (i>=0) ? (src2 >> i) : (src2 << -i) ;

               setcondcode(result) ;

              end

    `ROT    : begin

               clearcondcode ;

               src1 = getsrc(ir) ;

               src2 = getdst(ir) ;

               dir = (src1[ADDRSIZE-1]==0) ? `RIGHT : `LEFT ;

               i = (src1[ADDRSIZE-1]==0) ?

                   src1 : -src1[ADDRSIZE-1:0];

               while (i > 0) begin

                  if (dir == `RIGHT) begin

                     result = src2 >> 1 ;

                     result[WIDTH-1] = src2[0] ;

                  end

                  else begin

                  result = src2 << 1 ;

                      result[0] = src2[WIDTH-1] ;

                  end

                  i = i - 1 ;

                  src2 = result ;

               end

               setcondcode(result) ;

              end

    `HLT    : begin

                $display("Halt ...") ;

                $stop ;

              end

    default     : $display("Error : Illegal Opcode .") ;

  endcase

end

endtask

// Запись результата в регистровый файл или в память

task write_result ;

begin

  if ((`OPCODE >= `ADD) && (`OPCODE < `HLT)) begin

     if (`DSTTYPE == `REGTYPE) RFILE[`DST] = result ;

     else MEM[`DST] = result ;

  end

end

endtask

//  Помощь при отладке ....

task apply_reset ;

begin

  reset = 1 ;

  #CYCLE

  reset = 0 ;

  pc = 0 ;

end

endtask

task disprm ;

input rm ;

input [ADDRSIZE-1:0] adr1, adr2 ;

begin

  if (rm == `REGTYPE) begin

     while (adr2 >= adr1) begin

        $display("REGFILE[%d]=%d\n",adr1,RFILE[adr1]) ;

        adr1 = adr1 + 1 ;

     end

  end

  else begin

     while (adr2 >= adr1) begin

        $display("MEM[%d]=%d\n",adr1,MEM[adr1]) ;

        adr1 = adr1 + 1 ;

     end

  end

end

endtask

// Блоки initial и always

initial begin : prog_load

    $readmemb("sisc.prog",MEM) ;

    $monitor("%d %d  %h  %h  %h",

              $time,pc,RFILE[0],RFILE[1],RFILE[2]) ;

    apply_reset ;

end

always begin : main_process

  if (!reset) begin

     #CYCLE fetch ;

     #CYCLE execute ;

     #CYCLE write_result ;

  end

  else #CYCLE ;

end

endmodule

/*===========================================

 *

 *

 * Модель SISC процессора с конвейером

 *

 *sisc_pipeline_model.v

 *

 * %W%  %G%  -- Для управления версией

*/

module pipeline_control ;

// Здесь должны быть включены все описания, функции и

//задачи, определенные ранее. Кроме того, следующие

//описания и объявления используются только для

//моделирования управления конвейера.

// Объявление параметров

parameter   HALFCYCLE = (CYCLE/2) ; //Половина времени такта

parameter   QDEPTH = 3 ; //Длина очереди команд

// Объявление дополнительных регистров для

// управления конвейером

reg [WIDTH-1:0]  IR_Queue[0:QDEPTH-1], //Очередь команд

                 wir,     //Регистр команды для уровня

                          //записи результата

reg [2:0]        eptr, fptr, qsize ;// Указатели

reg              clock ;  //Системный генератор

                          //синхроимпульсов

reg [WIDTH:0]    wresult; //Регистр результата АЛУ

                          //для уровня записи результата

// Различные признаки - линии управления

reg     mem_access, branch_taken, halt_found ;

reg     result_ready ;

reg     executed, fetched ;

reg     debug ;

wire        queue_full ;

event       do_fetch, do_execute, do_write_results ;

// Описание полей команды

`define WOPCODE  wir[31:28]

`define WDST     wir[11:0]

`define WDSTTYPE wir[26] // тип назначения, 0=регистр,

                         //1=imm

// Непрерывное присваивание для заполненной

// очереди (queue_full)

assign  queue_full = (qsize == QDEPTH) ;

// Задачи и функции

task fetch ;

begin

  IR_Queue[fptr] = MEM[pc] ;

  fetched = 1 ;

end

endtask

task execute ;

begin

  if (!mem_access) ir = IR_Queue[eptr] ;//Нужно новое
                                        //содержимое

                                        //регистра команд

                                        //(IR)

  case (`OPCODE)

    `NOP    : begin

                if (debug) $display("Nop ...") ;

              end

    `BRA    : begin

                if (debug) $write("Branch ...") ;

                if (checkcond(`CCODE) == 1) begin

                   pc = `DST ;

                   branch_taken = 1 ;

                end

              end

    `LD     : begin

              if (mem_access == 0) begin

                 mem_access = 1 ; //Резервирование

                                  //следующего такта

              end

              else begin       // Доступ к памяти

               if (debug) $display("Load ...") ;

                   clearcondcode ;

               if (`SRCTYPE) begin

                  RFILE[`DST] = `SRC ;

               end

               else RFILE[`DST] = MEM[`SRC] ;

                   setcondcode({1'b0,RFILE[`DST]}) ;

                   mem_access = 0 ;

               end

              end

    `STR    : begin

                if (mem_access == 0) begin

                   mem_access = 1 ; //Резервирование

                                    //следующего такта

                end

                else begin      //Доступ к памяти

                   if (debug) $display("Store ...") ;

                   clearcondcode ;

                   if (`SRCTYPE) begin

                      MEM[`DST] = `SRC ;

                   end

                   else MEM[`DST] = RFILE[`SRC] ;

                      mem_access = 0 ;

                end

              end

 // Команды ADD, MUL, CMP, SHF и ROT

 // моделируются так же, как и в первой модели.

    `HLT    : begin

                $display("Halt ...") ;

                halt_found = 1 ;

              end

    default     : $display("Error : Wrong Opcode in instruction.") ;

    endcase

   if (!mem_access) executed = 1 ; // Команда выполнена?

end

endtask

task write_result ;

begin

  if ((`WOPCODE >= `ADD) && (`WOPCODE < `HLT)) begin

     if (`WDSTTYPE == `REGTYPE) RFILE[`WDST] = wresult ;

     else MEM[`WDST] = wresult ;

     result_ready = 0 ;

  end

end

endtask

task flush_queue ;

begin

  // Счетчик команд pc уже изменен при выполнении

  //команды branch

  fptr = 0 ;

  eptr = 0 ;

  qsize = 0 ;

  branch_taken = 0 ;

end

endtask

task copy_results ;

begin

  if ((`OPCODE >= `ADD) && (`OPCODE < `HLT)) begin

     setcondcode(result) ;

     wresult = result ;

     wir = ir ;

     result_ready = 1 ;

  end

end

endtask

task set_pointers ; //Управление указателями очереди

begin

   case ({fetched,executed})

    2'b00 : ;

    2'b01 : begin      // Нет выборки

               qsize = qsize - 1 ;

               eptr = (eptr + 1)%QDEPTH ;

            end

    2'b10 : begin      // Нет выполнения

               qsize = qsize + 1 ;

               fptr = (fptr + 1) % QDEPTH ;

            end

    2'b11 : begin      //Выборка и выполнение

               eptr = (eptr + 1)%QDEPTH ;

               fptr = (fptr + 1) % QDEPTH ;

            end

   endcase

end

endtask

// Помощь при отладке ....

task disprm ;

input rm ;

input [ADDRSIZE-1:0] adr1, adr2 ;

begin

  if (rm == `REGTYPE) begin

     while (adr2 >= adr1) begin

        $display("REGFILE[%d]=%d\n",adr1,RFILE[adr1]) ;

        adr1 = adr1 + 1 ;

     end

  end

  else begin

     while (adr2 >= adr1) begin

        $display("MEM[%d]=%d\n",adr1,MEM[adr1]) ;

        adr1 = adr1 + 1 ;

     end

  end

end

endtask

task  apply_reset;   //Сброс системы
begin

  reset = 1 ;

  @(negedge clock) #1 ;

  reset = 0 ;

  pc = 0 ;

  mem_access = 0 ;

  branch_taken = 0 ;

  halt_found = 0 ;

  qsize = 0 ;

  result_ready = 0 ;

  eptr = 0 ;

  fptr = 0 ;

  executed = 0 ;

  fetched = 0 ;

end

endtask

//Блоки Initial и Always ...

initial begin : setup_and_prog_load

    `DEBUG_OFF ;

    waves ;

    $readmemb("rsisc.prog",MEM) ;

    apply_reset ;

end

initial begin : clock_loop  //Цикл  clock

  clock = 1 ;

  forever begin

    #(HALFCYCLE) clock = ~clock ;

  end

end

always @(posedge clock) begin : phase1_loop

  if (!reset) begin

     fetched = 0 ; executed = 0 ;

     if (!queue_full && !mem_access) -> do_fetch ;

     if (qsize || mem_access)        -> do_execute ;

     if (result_ready)               -> do_write_results;

  end

end

always @(negedge clock) begin : phase2_block

  if (!reset) begin //Копирование результата, установка

                    // кодов условий
     if (!mem_access && !branch_taken)

        copy_results ;

     if (branch_taken) pc = `DST ;//Установить следующее
                                  //значение счетчика

                                  //команд (PC)

     else if (!mem_access) pc = pc+1; //Не изменять
                                      //при пропуске такта...

     if (branch_taken || halt_found)  //Очистка очереди
         flush_queue ;

     else set_pointers ; //или корректировка указателей

     if (halt_found) begin //Команда  Halt?

         $stop ;

         halt_found = 0 ;  // Сбросить halt

     end

  end

end

// Эти события управляют тремя уровнями конвейера

// Уровень выборки
always @(do_fetch) begin : fetch_block

       fetch ;

end

// Уровень выполнения

always @(do_execute) begin : execute_block

  execute ;

end

// Уровень записи
always @(do_write_results) begin : write_block

     write_result ;

end

endmodule

     Рис. 3.22 Модель SISC процессора с конвейерной

               обработкой на языке VERILOG

